



TERMINAL

Financiële programma's voor MSX Computers

Robert-Jan Donkers

Financiële Praktijkprogramma's

- Beleggen
- Sparen
- Lenen
- Bedrijfsanalyse
- Vaste Lasten Bewaking
- enz. enz.

Terminal Software Publicaties

FINANCIËLE PROGRAMMA'S
voor MSX COMPUTERS
ROBERT-JAN DONKERS

TERMINAL SOFTWARE PUBLICATIES

Een uitgave van: Terminal Software Publicaties
Postbus 111, 5110 AC Baarle-Nassau

1e druk juni 1985

(c) 1985, Robert-Jan Donkers

ISBN 90-6883-011-2

CIP-GEGEVENS KONINKLIJKE BIBLIOTHEEK, DEN HAAG

Donkers, R.-J.

Financiële programma's voor MSX Computers / Robert-Jan Donkers. - Baarle-Nassau: Terminal Software Publicaties
ISBN 90-6883-011-2
SISO 365.3 UDC 681.3.06
Trefw.: computerprogramma's / MSX Computer (computer).

Niets uit deze uitgave mag worden veeelvoudigd en/of openbaar gemaakt worden door middel van druk, fotokopie, microfilm, elektronische informatiedragers of op welke andere wijze ook, zonder voorafgaande schriftelijke toestemming van de uitgever.

Ondanks alle zorg waarmee deze uitgave is samengesteld kan noch de auteur noch de uitgever enige aansprakelijkheid aanvaarden voor eventuele schade die zou kunnen voortvloeien uit het gebruik van deze programma's of uit enig andere fout die in deze uitgave zou kunnen voorkomen. MSX is een handelsmerk van MICROSOFT INC.

VOORWOORD

Met deze verzameling financiële programma's hebben we getracht voor de gebruiker van de homecomputer een aantal voorbeelden te geven hoe de homecomputer voor serieuze zaken te benutten.

U vindt in dit boek een aantal kleine programma's, waarmee het mogelijk is de meestal ingewikkelde financiële berekeningen snel en efficiënt uit te voeren. Zo hebben we berekening opgenomen i.v.m. beleggen, lenen, sparen en afschrijven. De berekeningen worden allemaal voorafgegaan door de Formule waarmee gerekend wordt en een praktisch voorbeeld.

Naast deze berekeningen hebben we een aantal financiële programma's opgenomen die geheel op zichzelf staan, o.a. een HYPOTHEEK-berekening, BREAK-EVEN POINT ANALYSE, VASTE LASTEN BEWAKING, een simpel doch efficiënt FAKTUREERPROGRAMMA, een programma om PRIJSLIJSTEN te maken en een programma om FINANCIËLE KENGETALLEN voor een bedrijf te berekenen. Deze laatste zijn stuk voor stuk zonder meer in een bedrijf(je) toe te passen.

Tenslotte zouden we nog willen voorstellen om de diverse berekeningen in EEN programma op te nemen zodat U alle soorten berekeningen steeds in een programma bij de hand heeft. In dit boek hebben we echter voor de duidelijkheid alle berekeningen in aparte programma's opgenomen.

Wij hopen dat U in de praktijk veel plezier van deze programma's zult hebben.

juni 1985,
de auteur.

INHOUD

	Pag.
BELEGGEN	
Berekeningen 1 t/m 6	9
SPAREN	
Berekeningen 1 en 2	21
RENTETABEL	25
LENEN	
Berekeningen 1 t/m 6	31
HYPOTHEEK AFLOSSINGSTABEL	46
AFSCHRIJVEN	
Berekeningen 1 t/m 6	54
ECONOMISCHE BESTELHOEVEELHEID	61
BREAK-EVEN POINT BEREKENING	65
EENHEIDSPRIJSBEREKENING	67
VASTE LASTEN BEWAKING	70
FAKTUREREN	79
PRIJSLIJSTEN	88
FINANCIËLE BEDRIJFSANALYSE	97

BELEGGEN

Investerings 1: Toekomstige waarde van een belegging

De toekomstige waarde van een belegging wordt berekend met de formule:

$$T = P \times ((1 + I/N)^N \times J)$$

waarbij: T = totale waarde na J Jaren beleggen

P = Oorspronkelijke belegging

I = De nominale Rente

N = Aantal rentebijschrijvingen/jaar

J = Aantal jaren dat de belegging uitstaat

Bij dit programma wordt er van uitgegaan, dat er tussentijds geen opnames of stortingen plaatsvinden.

Voorbeeld:

Piet doet een investering van f 5000,00 tegen een nominale rente van 10%. Wat zal zijn investering totaal waard zijn, als hij dit bedrag gedurende vijf jaren uitzet en de rente per kwartaal wordt bijgeschreven?

TOEKOMSTIGE INVESTERINGS WAARDE.

Aanvangs investering : 5000

Nominaal rentepct. : 10

Aantal rentebijschrijvingen
per jaar : 4

Looptijd in jaren : 5

Toekomstige waarde = Fl. 8193.13

Nog een berekening ? (Ja of Nee)


```

10 REM *****
    **   INVESTERINGEN 1   **
    *****
20 CLS:COLOR 15,1,1:WIDTH 40:KEYOFF
30 PRINT"TOEKOMSTIGE INVESTERINGS WAARDE
." :PRINT
40 PRINT"%%%%%%%%%%%%%%
%%%%%%%%%"
50 PRINT
60 INPUT"AANVANGSINVESTERING      ":";P
70 PRINT
80 INPUT"NOMINAAL RENTEPCT.      ":";I
90 PRINT
100 INPUT"AANTAL RENTEBIJSCHRIJVINGEN
      PER JAAR                      ":";N
110 PRINT
120 INPUT"LOOPTIJD IN JAREN      ":";J
130 PRINT:PRINT"%%%%%%%%%%%%%%
%%%%%%%%%"
140 PRINT
150 REM *****
    **   BEREKEN DE RENTE PER   **
    **   PERIODE                 **
    *****
160 I=I/N/100
170 REM *****
    **   BEREKEN NU DE         **
    **   DE TOEKOMSTIGE WAARDE  **
    **   D.M.V. DE FORMULE     **
    *****
x 180 T=P*(1+I)^(N*J)
190 PRINT"TOEKOMSTIGE WAARDE =F1. ";INT(T
*100+.5)/100
210 LOCATE 0,21:PRINT"Nog een berekening
Ja/Nee ? "
220 Q$=INKEY$:IF Q$="" THEN 220
230 IF Q$="J" OR Q$="j" THEN RUN
240 IF Q$<>"N" AND Q$<>"n" THEN 220
250 END

```

Investerings 2: Regelmatige opnames uit een belegging

Met dit programma berekent U welk bedrag U maandelijks kunt opnemen van een belegging gedurende een aantal jaren tot het restbedrag van de belegging NUL is.

De formule voor deze berekening is:

$$R = Px((I/N)/((1 + I/N)^{(N \times J) - 1}) + (I/N))$$

waarbij: R = Bedrag van de regelmatige opname

P = Basisbedrag van de belegging

I = De nominale intrest

N = Aantal malen dat het bedrag R wordt opgenomen per jaar

J = aantal jaren dat het basisbedrag P uitstaat

Voorbeeld:

De vader van Pieter besluit het wat rustiger aan te doen en heeft daarom besloten zijn aandeel in de zaak van zijn broer te verkopen.

Zijn broer besluit het zelf te kopen voor het lieve sommetje van f 125.000,00.

Omdat hij nog geen recht op AOW heeft, besluit hij het bedrag in de 10 jaren tot zijn pensioengerechtigde leeftijd op te souperen. De rentevergoeding bedraagt 9,5%. Wat kan de vader van Pieter nu maandelijks opnemen gedurende die 10 jaren, zodat zijn gehele belegging na 10 jaar precies NUL is?

REGELMATIGE OPNAMES
UIT EEN BELEGGING

Basis belegging	:125000
Nominale rente pct.	:9.5
Aantal opnames (p/j)	:12
Looptijd	:10

BEDRAG PER OPNAME=FL.1617.47

Nog een berekening ? (Ma of Nee)

```

10 REM *****
    **      INVESTERINGEN 2      **
    *****
20 CLS:COLOR 15,1,1:WIDTH 40:KEYOFF
30 PRINT"          REGELMATIGE OPNAMES
          UIT EEN BELEGGING":P
RINT
40 PRINT"%%%%%%%%%"
    %%%%%%%%%%"
50 PRINT
60 INPUT"BASIS BELEGGING          ":";P
70 PRINT
80 INPUT"NOMINAAL RENTEPCT.        ":";I
90 PRINT
100 INPUT"AANTAL OPNAMES  P/J      ":";N
110 PRINT
120 INPUT"LOOPTIJD IN JAREN        ":";J
130 PRINT:PRINT"%%%%%%%%%"
    %%%%%%%%%%"
140 PRINT
150 REM *****
    ** BEREKEN DE RENTE PER      **
    **      PERIODE              **
    *****
160 I=I/N/100
170 REM *****
    **      BEREKEN NU DE        **
    ** DE REGELMATIGE OPNAME     **
    ** D.M.V. DE FORMULE        **
    *****
180 R=P*(I/((1+I)^(N*J)-1))+I)
190 PRINT"BEDRAG PER OPNAME  =F1.";
195 PRINT USING"#####.##";R
210 LOCATE 0,21:PRINT"Nog een berekening
    Ja/Nee ? "
220 Q$=INKEY$:IF Q$="" THEN 220
230 IF Q$="J" OR Q$="j" THEN RUN
240 IF Q$<>"N" AND Q$<>"n" THEN 220
250 END

```

Investerings 3: Berekenen van basisbelegging om een bepaald streefbedrag te verkrijgen

Dit programma berekent voor U wat U als basisbedrag moet beleggen om na een J aantal jaren een bepaald bedrag bij elkaar te hebben.

De berekening is volgens de formule: $P = T / ((1 + I/N)^{N \times J})$

waarbij: P = Basisbelegging

T = Gewenst Eindbedrag

N = Aantal rentebijschrijvingen per jaar

J = Aantal jaren dat U wilt beleggen om het bedrag T bij elkaar te krijgen

I = De nominale intrest

Voorbeeld:

Hoeveel moet U beleggen tegen een rente van 8,5% om na 10 jaar een bedrag van f 20.000,00 bij elkaar te hebben vergaard? De rente wordt per kwartaal bijgeschreven.

Uiteraard wordt hier rekening gehouden met rente op rente.

```

BEREKENEN VAN BASISBELEGGING
=====
Streefbedrag                : 200000
Aantal rentebijdrifvingen  : 4
per jaar
Aantal jaren                 : 10
Nominale rente pct.         : 8.5
=====
Aanvangs investering=Fl.8624.76

Nog een berekening ? (Ja of Nee)

```

```

10 REM *****
    **      INVESTERINGEN 3      **
    *****
20 CLS:COLOR 15,1,1:WIDTH 40:KEYOFF
30 PRINT"      BEREKENEN VAN BASISBELEGGING
";PRINT
40 PRINT"%%%%%%%%%"
"%%%%%%%%%"
50 PRINT
60 INPUT"STREEFBEDRAG              ":";T
70 PRINT
80 INPUT"AANTAL RENTEBIJSCHRIJVINGEN
PER JAAR                          ":";N
90 PRINT
100 INPUT"AANTAL JAREN              ":";J
110 PRINT
120 INPUT"NOMINAAL RENTE PERCT.    ":";I
130 PRINT:PRINT"%%%%%%%%%"
"%%%%%%%%%"
140 PRINT
150 REM *****
    ** BEREKEN DE RENTE PER      **
    **      PERIODE              **
    *****
160 I=I/N/100
170 REM *****
    **      BEREKEN NU DE        **
    ** AANVANGSINVESTERING      **
    ** D.M.V. DE FORMULE        **
    *****
180 P=T/(1+I)^(N*J)
190 PRINT"AANVANGSINVESTERING =F1.";
195 PRINT USING"#####.##";P
210 LOCATE 0,21:PRINT"Nog een berekening
Ja/Nee ? "
220 Q$=INKEY$:IF Q$="" THEN 220
230 IF Q$="J" OR Q$="j" THEN RUN
240 IF Q$<>"N" AND Q$<>"n" THEN 220
250 END

```

Investerings 4: Berekenen van basisbelegging om regelmatige opnames te kunnen doen

Dit programma berekent welk bedrag U minimaal moet investeren om gedurende een aantal jaren bepaalde bedragen te kunnen opnemen uit Uw belegging. Het programma gaat er vanuit dat dit een minimum investering is, dit betekent dat ervan wordt uitgegaan dat het restbedrag na die jaren NUL is. Indien U meer stort dan nodig is voor Uw regelmatige opnames heeft U nog een aardige som over na die jaren.

De berekening geschiedt met de formule:

$$P = (R \times N / I) \times (((1 - (1 / ((1 + I / N)^{(N \times J)}))))$$

waarbij: P = Minimale basisbelegging

R = Bedrag van de regelmatige opname

I = De nominale intrest

N = Aantal opnames per jaar

J = Aantal jaren dat U regelmatige opnames wilt doen

Voorbeeld:

Hoeveel moet ik minimaal investeren om gedurende 12 jaar elke maand f 980,00 op te nemen, als de rentevergoeding 6,5% is?

MINIMALE BELEGGING
VOOR PERIODIEKE OPNAMES

Bedrag van opnames : 980

Nominale rente pct. : 6.5

Aantal opnames (p/j) : 12

Aantal jaren : 12

Min. investering = Fl.97812.09

Nog een berekening ? (Ja of Nee)

```

10 REM *****
    **   INVESTERINGEN 4   **
    *****
20 CLS:COLOR 15,1,1:WIDTH 40:KEYOFF
30 PRINT "                MINIMALE BELEGGING
                        VOOR PERIODIEKE OPNAME
S":PRINT
40 PRINT "%%%%%%%%%"
    %%%%%%%%%"
50 PRINT
60 INPUT "BEDRAG VAN OPNAMES      ":";R
70 PRINT
80 INPUT "NOMINAAL RENTEPCT.      ":";I
90 PRINT
100 INPUT "AANTAL OPNAMES P/J      ":";N
110 PRINT
120 INPUT "AANTAL JAREN            ":";J
130 PRINT:PRINT "%%%%%%%%%"
    %%%%%%%%%"
140 PRINT
150 REM *****
    **   BEREKEN DE RENTE IN   **
    **   DECIMALEN             **
    *****
160 I=I/100
170 REM *****
    **   BEREKEN NU DE        **
    **   MINIMALE INVESTERING **
    **   D.M.V. DE FORMULE    **
    *****
180 P=R*N/I*(1-1/((1+I/N)^(N*J)))
190 PRINT "MIN. INVESTERING  = F1.";
195 PRINT USING "#####.##";P
210 LOCATE 0,21:PRINT "Nog een berekening
    Ja/Nee ? "
220 Q$=INKEY$:IF Q$="" THEN 220
230 IF Q$="J" OR Q$="j" THEN RUN
240 IF Q$<>"N" AND Q$<>"n" THEN 220
250 END

```

Investerings 5: Berekenen van de nominale rente op investeringen

Met dit programma berekent U de nominale rente van een investering als U van tevoren weet wat de investering waard is nadat het een aantal jaren heeft uitgestaan.

De formule voor deze berekening is:

$$I = (N \times (T/P) \mid (1/N \times J) - N$$

waarbij: I = De nominale intrest

P = Basisinvestering

T = De toekomstige waarde van de investering

N = Aantal rentebijgeschrijvingen per jaar

J = Aantal jaren dat de belegging uitstaat

Voorbeeld:

Jan besluit om f 3000,00 te beleggen. Na 5 jaar blijkt zijn tegoed op de beleggingsrekening f 4150,00 te zijn.

Wat is de nominale rentevergoeding die hij met deze belegging heeft gerealiseerd?

NOMINAAL RENTEPERCENTAGE
OP INVESTERINGEN

Aanvangs investering : 3000
Totale waarde : 4150
Aantal jaren : 5
Aantal rente bij-
schrijvingen per jaar : 4

Nominale rente = 6.5428557 %

Nog een berekening ? (Ja of Nee)


```

10 REM *****
    **      INVESTERINGEN 5      **
    *****
20 CLS:COLOR 15,1,1:WIDTH 40:KEYOFF
30 PRINT"          NOMINAAL RENTEPERCENTAG
E          OP INVESTERINGEN":P
RINT
40 PRINT"%%%%%%%%%%%%%%
%%%%%%%%%"
50 PRINT
60 INPUT"AANVANGSINVESTERING      ":"P
70 PRINT
80 INPUT"TOTALE EINDWAARDE        ":"T
90 PRINT
100 INPUT"AANTAL JAREN              ":"J
110 PRINT
120 INPUT"AANTAL RENTEBIJSCHRIJVINGEN
    PER JAAR                        ":"N
130 PRINT:PRINT"%%%%%%%%%%%%%%
%%%%%%%%%"
140 PRINT
170 REM *****
    **      BEREKEN NU HET      **
    **NOMINAAL RENTEPERCENTAGE**
    **      D.M.V. DE FORMULE   **
    *****
180 I=N*((T/P)^(1/(N*J))-1)*100
190 PRINT"NOMINALE RENTE      =      ";
195 PRINT USING"#####.##";I
210 LOCATE 0,21:PRINT"Nog een berekening
Ja/Neen ? "
220 Q$=INKEY$:IF Q$="" THEN 220
230 IF Q$="J" OR Q$="j" THEN RUN
240 IF Q$<>"N" AND Q$<>"n" THEN 220
250 END

```

Investerings 6: Berekenen van effectieve rente op een investering

Met dit programma dat direct aansluit op INVESTERINGEN 5 berekent U de effectieve rente van een bepaalde investering.

De berekening van de effectieve rente geschiedt met de formule:

$$E = ((T/P) \mid (1/J) - 1) \times 100$$

waarbij: E = Effectief rentepercentage

P = Basisinvestering

T = Waarde van investering na J jaren

J = Aantal jaren dat de belegging uitgestaan heeft

Voorbeeld:

Jan heeft f 12.000,00 belegd, na 5 jaren was zijn belegging f 14.900,00 waard.
Wat is het effectieve rentepercentage van zijn belegging?

EFFECTIEF RENTEPERCENTAGE
OP INVESTERINGEN

Aanvangs investering :12000
Totale eind-waarde :14900
Aantal jaren :5

Rente pct. (p/j) = 4.4241634 %

Nog een berekening ? (Ja of Nee)

```

10 REM *****
    **      INVESTERINGEN 6      **
    *****
20 CLS:COLOR 15,1,1:WIDTH 40:KEYOFF
30 PRINT"          EFFECTIEF RENTEPERCENTAG
E          OP INVESTERINGEN":P
RINT
40 PRINT"%%%%%%%%%"
    %%%%%%%%%"
50 PRINT
60 INPUT"AANVANGSINVESTERING      ":";P
70 PRINT
80 INPUT"TOTALE EINDWAARDE        ":";T
90 PRINT
100 INPUT"AANTAL JAREN              ":";J
110 PRINT
130 PRINT:PRINT"%%%%%%%%%"
    %%%%%%%%%"
140 PRINT
170 REM *****
    **      BEREKEN NU HET      **
    **EFFECTIEF RENTEPERCENTAGE**
    **      D.M.V. DE FORMULE   **
    *****
180 I=((T/P)^(1/J)-1)*100
190 PRINT"EFFECTIEVE RENTE      =      ";
195 PRINT USING"#####.##";I
210 LOCATE 0,21:PRINT"Nog een berekening
Ja/Nee ? "
220 Q$=INKEY$:IF Q$="" THEN 220
230 IF Q$="J" OR Q$="j" THEN RUN
240 IF Q$<>"N" AND Q$<>"n" THEN 220
250 END

```

Sparen 1: Toekomstige waarde berekenen van regelmatig sparen

De toekomstige waarde van zo'n spaarvorm wordt berekend met de formule:

$$T = R \times (((1 + I/N)^{(N \times J)} - 1) / (I/N))$$

waarbij: T = Totale waarde na J jaren sparen

R = Het regelmatige spaarbedrag

N = Aantal stortingen per jaar

J = Aantal jaren dat er gespaard wordt

I = Nominale Rente

Bij dit programma wordt er van uitgegaan, dat alle stortingen gelijk zijn en er tussentijds geen opnames plaatsvinden. Bovendien wordt verondersteld dat de rente wordt bijgeschreven elke keer als er gestort wordt.

Voorbeeld:

Pieter besluit elke maand f 50,00 te sparen tegen een rente van 5%. Hoeveel heeft Pieter bij elkaar gespaard na 5 jaar?

TOEKOMSTIGE WAARDEN VAN
REGELMATIGE DEPOSITO'S

Bedrag van deposito : 50

Rente percentage : 5

Aantal deposito's (p/J) : 12

Aantal jaren : 5

Toekomstige waarde = Fl. 3400.3

Nog een berekening? (Ja of Nee)

```

10 REM *****
    **          SPAREN 1          **
    *****
20 CLS:COLOR 15,1,1:WIDTH 40:KEYOFF
30 PRINT"          TOEKOMSTIGE WAARDE VAN
          REGELMATIGE DEPOSITO'S

":PRINT
40 PRINT"%%%%%%%%%%%%%%
%%%%%%%%%"
50 PRINT
60 INPUT"BEDRAG VAN DEPOSITO      ":";R
70 PRINT
80 INPUT"RENTEPERCENTAGE          ":";I
90 PRINT
100 INPUT"AANTAL DEPOSITO'S P/J ":";N
110 PRINT
120 INPUT"AANTAL JAREN            ":";J
130 PRINT:PRINT"%%%%%%%%%%%%%%
%%%%%%%%%"
140 PRINT
150 REM *****
    **  BEREKEN HET RENTEPCT.  **
    **    PER DEPOSITO        **
    **VAN PROCENT NAAR DECIMAAL**
    *****

160 I=I/N/100
170 REM *****
    **    BEREKEN NU DE        **
    **  TOEKOMSTIGE WAARDE    **
    **    D.M.V. DE FORMULE    **
    *****

180 T=R*((1+I)^(N*J)-1)/I
190 PRINT"TOEKOMSTIGE WAARDE = F1.";
195 PRINT USING"#####.##";T
210 LOCATE 0,21:PRINT"Nog een berekening
    Ja/Nee ? "
220 Q$=INKEY$:IF Q$="" THEN 220
230 IF Q$="J" OR Q$="j" THEN RUN
240 IF Q$<>"N" AND Q$<>"n" THEN 220
250 END

```

Sparen 2: Sparen met de bedoeling een bepaald eindbedrag te vergaren

Met sparen 2 kunt U berekenen welk bedrag U maandelijks moet sparen om na een bepaalde tijd een streefbedrag bij elkaar te hebben vergaard.

De berekening geschiedt met de formule:

$$R = T \times (((I/N)/((1+I/N)^{(N \times J)} - 1)))$$

waarbij: T = Streefbedrag

R = Bedrag dat men regelmatig moet sparen

I = De nominale Intrest

N = Aantal stortingen per jaar

J = Aantal jaren waarin men het streefbedrag bij elkaar wilt hebben

Voorbeeld:

Pieter besluit een nieuwe printer voor zijn homecomputer te kopen. Dit apparaat kost f 1500,00 en hij wilt het in 1 jaar bij elkaar gespaard hebben. Hij spaart elke maand en de rente op zijn spaarrekening is 6,5%. Welk bedrag moet Pieter nu elke maand sparen om na 1 jaar precies f 1500,00 te hebben gespaard?

REGELMATIGE DEPOSITO'S

Totale eindwaarde :1500

Nominale rente pct. :6.5

Aantal deposito's (p/j):12

Aantal jaren :1

Regelm. deposito's =Fl.111.68

Nog een berekening ? (Ja of Nee)

```

10 REM *****
    **          SPAREN 2          **
    *****
20 CLS:COLOR 15,1,1:WIDTH 40:KEYOFF
30 PRINT"          REGELMATIGE DEPOSITO'S
":PRINT
40 PRINT"%%%%%%%%%"
"%%%%%%%%%"
50 PRINT
60 INPUT"STREEFBEDRAG          ":";T
70 PRINT
80 INPUT"RENTEPERCENTAGE          ":";I
90 PRINT
100 INPUT"AANTAL DEPOSITO'S P/J ":";N
110 PRINT
120 INPUT"AANTAL JAREN          ":";J
130 PRINT:PRINT"%%%%%%%%%"
"%%%%%%%%%"
140 PRINT
150 REM *****
    **  BEREKEN HET RENTEPCT.  **
    **    PER DEPOSITO    **
    **VAN PROCENT NAAR DECIMAAL**
    *****
160 I=I/N/100
170 REM *****
    **          BEREKEN NU HET          **
    **    REGELMATIGE DEPOSITO    **
    **    D.M.V. DE FORMULE    **
    *****
180 R=T*I/((I+1)^(N+J)-1)
190 PRINT"REGELM.DEPOSITO          = F1.";
195 PRINT USING"#####.###";R
210 LOCATE 0,21:PRINT"Nog een berekening
Ja/Nee ? "
220 Q$=INKEY$:IF Q$="" THEN 220
230 IF Q$="J" OR Q$="j" THEN RUN
240 IF Q$<>"N" AND Q$<>"n" THEN 220
250 END

```

RENTETABEL

Met dit programma krijgt U een overzicht van de verdiende rente in tabelvorm. U krijgt daarmee een duidelijk overzicht van de renteontwikkeling van Uw investering.

In de tabel vindt U de volgende gegevens

- a) Jaar van overzicht
- b) Balans van Uw belegging
- c) Rente tussen 2 opeenvolgende perioden
- d) Cumulatieve rente

Daarnaast wordt voor U het effectief rentepercentage berekend.

Voorbeeld:

Karel heeft f 1500,00 uitstaan tegen een rente van 6%. De rente wordt per kwartaal bijgeschreven. Wat is de renteontwikkeling van deze belegging in het zesde en zevende jaar.

VERDIENDE RENTE TABEL.			
Aanvangs investering: Fl. 1500			
Nominiaal rente pct. : 6 %			
Looptijd : 7 jaar.			
Effectief rente pct. : 6.14%(p/j)			
JAAR	BALANS	RENTE	CUM.RENTE
6	2050.58	550.58	550.58
	2081.34	30.76	581.34
	2112.56	31.22	612.56
	2144.25	31.69	644.25
7	2176.41	32.16	676.41
	2209.06	32.65	709.06
	2242.2	33.14	742.2
	2275.83	33.63	775.83
Nog een berekening ? (Ja of Nee)			


```

10  REM *****
    **
    **   RENTE - TABEL   **
    **
    *****
20  CLS:WIDTH 40:COLOR 15,1,1:KEYOFF
30  DEF FNQ(X)=INT(X*100+.5)/100
40  PRINT"          VERDIENDE RENTE BEREKENE
N":PRINT
50  PRINT"%%%%%%%%%"
    %%%%%%%%%":PRINT
60  INPUT "AANVANGSINVESTERING: ";P
70  PRINT
80  INPUT "NOMINAAL RENTEPCT. : ";I
90  REM *****
    ** VAN PROCENT NAAR **
    **   DECIMAAL       **
    *****
100 I=I/100:PRINT
110 INPUT"AANTAL DEPOSITO'S OF
    OPNAMES PER JAAR : ";N1
120 REM *****
    ** VRAAG NIET NAAR **
    **   BEDRAG ALS N1=0 **
    *****
130 IF N1=0 THEN 230
140 REM *****
    ** DEPOSITO'S WORDEN **
    ** INGEVOERD ALS EEN **
    ** POSITIEF GETAL    **
    *****
150 REM *****
    ** OPNAMES   WORDEN **
    ** INGEVOERD ALS EEN **
    ** NEGATIEF GETAL    **
    *****
160 PRINT
170 INPUT"BEDRAG VAN STORTING(+)
    OF OPNAME(-)      : ";R

```

```

180 REM *****
    ** RENTE WORDT DAGELIJKS **
    **   BIJGESCHREVEN       **
    *****
190 N=360
200 REM *****
    ** PRINTEN VAN DEPOSITO'S**
    **   EN OPNAMES         **
    *****
210 L2=N1
220 GOTO 280
230 PRINT
240 INPUT "AANTAL RENTE BIJ-
        SCHRIJVINGEN PER JAAR:";N
250 N1=0
260 REM *****
    ** PRINT DE GEGEVENS 4x  **
    **   PER JAAR           **
    *****
270 L2=4
280 PRINT:INPUT "BEGINJAAR          ":X
290 PRINT:INPUT "LAATSTE JAAR       ":Y
300 REM *****
    ** START PRINTEN VANAF  **
    ** HET BEGIN VAN EEN JAAR**
    *****
310 X=INT(X)
320 REM *****
    ** INITIALISEREN VAN DE  **
    **   HULPVARIABLEN      **
    *****
330 B0=P
340 I1=0:I2=0:I3=0
350 K=0
360 Q=10:IF N1=0 THEN Q=7
370 P1=4
380 FOR J=1 TO INT(Y)+1
390 REM ** PRINTEN STARTEN ? **
400 IF J<X THEN 600
410 IF K=1 THEN 590

```

```

420 CLS
430 PRINT"          VERDIENDE RENTE-TABEL"
440 PRINT
450 PRINT"AANVANGSINVESTERING:F1. ";P
460 PRINT"NOMINAAL RENTEPCT.      ";I*10
0;" %"
470 PRINT"LOOPTIJD                  ":"Y;"
JAAR."
480 K=1
490 REM *****
    ** SLA HET VOLGENDE OVER **
    **ALS ER GEEN REGELMATIGE**
    ** DEPOSITO'S OF OPNAMES **
    ** ZIJN.                  **
    *****
500 IF N1=0 THEN 530
510 PRINT"REGELMATIGE DEPOSITO'S
      OF OPNAMES          :F1. ";R
520 PRINT"AANTAL MALEN (P/J)      ":"N1
530 PRINT"EFFECTIEF RENTEPCT.    ":"FN Q
(100*((1+I/N)^N-1));" %(P/J)"
540 LOCATE 0,Q:PRINT"JAAR  BALANS      R
ENTE  CUM.RENTE"
550 GOTO 590
560 REM *** RENTE BEREKENING ***
570 PRINT
580 REM *** PRINT HET JAAR ***
590 PRINT J;
600 L1=1:N2=1:P2=1
610 FOR B=1 TO N
620 REM *****
    ** ZIJN ER NOG MEER DEPO-**
    ** SITO'S OF OPNAMES IN  **
    ** DIT JAAR ?          **
    *****
630 IF N2>N1 THEN 690
640 IF N2/N1>B/N THEN 690
650 REM *****
    ** BEREKEN NIEUWE BALANS **
    *****

```

```

660 B0=B0+R
670 REM *****
      ** TEL AANTAL OPNAMES OF **
      **   DEPOSITO'S           **
      *****
680 N2=N2+1
690 B2=B0*(1+I/N)
700 REM ** I1 IS RENTE PER      **
      **   BIJSCHRIJVING      **
710 I1=B2-B0
720 REM ** I3 IS DE GEACCUMULEERDE **
      **   RENTE TUSSEN DE     **
      **   BIJSCHRIJVINGEN     **
730 I3=I3+I1
740 REM ** I2 IS DE TOTAAL      **
      **   GEACCUMULEERDE RENTE **
750 I2=I2+I1
760 REM ** AFRONDEN OP BOEKINGSMOMENT**
770 IF P2/P1>B/N THEN 820
780 I2=FN Q(I2)
790 B2=FN Q(B2)
800 P2=P2+1
810 REM ** VANAF WELK JAAR PRINTEN **
820 IF J<X THEN 920
830 REM **   MOET ER EEN REGEL      **
      **   GEPRINT WORDEN ?        **
840 IF B/N<L1/L2 THEN 920
850 L1=L1+1
860 PRINT TAB(5); USING"#####.##";FN Q(
B2);:PRINT TAB(16);USING"#####.##";FN Q(
I3);:PRINT TAB(25);USING"#####.##"; FN Q
(I2)
870 Q=Q+1
880 IF Q>20 THEN GOSUB 1090
890 REM ** HULPVARIABELEN AANPASSEN **
900 I3=0
910 REM ** HET JAAR WORDT ALLEEN    **
      **   GEPRINT BIJ ELKE EERSTE **
      **   REGEL VAN EEN JAAR      **
920 B0=B2

```

```

930 REM ** GEEN REGELS MEER PRINTEN**
      ** VAN HET LAATSTE JAAR      **
940 IF J+B/N-1>=Y THEN 1030
950 NEXT B
960 REM ** PRINTEN ?? **
970 IF J<X THEN 1020
980 IF Q<21 AND N1<>0 AND Q<>10 THEN PRI
NT
990 IF Q<21 AND N1=0 AND Q<>7 THEN PRINT
1000 IF N1=0 AND Q<>7 OR N1<>0 AND Q<>10
  THEN Q=Q+1
1010 IF Q>20 AND J>=X THEN GOSUB 1090
1020 NEXT J
1030 PRINT
1040 LOCATE 0,21:PRINT"NOG EEN BEREKENIN
G ? (Ja/Nee)"
1050 Q$=INKEY$:IF Q$="" THEN 1050
1060 IF Q$="j" OR Q$="J" THEN RUN
1070 IF Q$="n" OR Q$="N" THEN END
1080 GOTO 1050
1090 LOCATE 0,23:PRINT" >RETURN< = VOLGE
NDE PAGINA "
1100 Q$=INKEY$:IF Q$="" THEN 1100
1110 IF ASC(Q$)<>13 THEN 1100
1120 Q=10
1130 IF N1=0 THEN Q=7
1140 CLS
1150 PRINT"          VERDIENDE RENTE-TABEL"
1160 PRINT
1170 PRINT"AANVANGSINVESTERING:F1. ";P
1180 PRINT"NOMINAAL RENTEPCT.      ";I*1
00;" %"
1190 PRINT"LOOPTIJD                  ":";Y;"
  JAAR."
1200 LOCATE 0,Q:PRINT"JAAR    BALANS
RENTE    CUM.RENTE"
1210 RETURN

```

LENINGEN

Leningen 1: Wat kan ik lenen?

Met dit programma berekent U snel wat er oorspronkelijk geleend is als U beschikt over de gegevens betreffende totale looptijd, jaarlijkse rente, aantal betalingen per jaar, bedrag dat regelmatig betaald moet worden.

De berekening geschiedt met de formule:

$$P = R \times N \times (1 - 1 / ((I / 100) / N + 1) \uparrow (N \times J)) / (I \times 100)$$

waarbij: R = Bedrag

I = Jaarlijkse rente

J = Looptijd van de lening

N = Aantal aflossingen per jaar

Voorbeeld:

Als ik maandelijks f 125,00 kan aflossen op een lening van 20% gedurende drie jaren. Wat is dan het bedrag, dat ik kan lenen?

HOEVEEL KAN IK LENEN ?

Bedrag van aflossing : 125

Aantal jaren : 3

Jaarlijks rentepct. : 20

Aantal aflossingen (p/J) : 12

Lening = Fl 3363.51

Nog een berekening ? (Ja of Nee)

```

10 REM *****
    **      LENINGEN      1      **
    *****
20 CLS:COLOR 15,1,1:WIDTH 40:KEYOFF
30 PRINT"      HOEVEEL KAN IK LENEN ?":
PRINT
40 PRINT"%%%%%%%%%%%%%%"
"%%%%%%%%%":PRINT
50 INPUT"BEDRAG VAN AFLOSSING  ":";R
60 PRINT
70 INPUT"AANTAL JAREN      ":";J
80 PRINT
90 INPUT"JAARLIJKS RENTEPCT.  ":";I
100 PRINT
110 INPUT"AANTAL AFLOSSINGEN P/J:";N
120 PRINT:PRINT"%%%%%%%%%%%%%%"
"%%%%%%%%%"
130 PRINT
150 REM *****
    **                                  **
    **      RENTE OMZETTEN VAN      **
    **      PROCENT NAAR DECIMAAL  **
    *****
160 I=I/100
170 REM *****
    **      BEREKEN NU HET      **
    **      TE LENEN BEDRAG      **
    **      D.M.V. DE FORMULE     **
    *****
180 P=R*N*(1-1/(I/N+1)^(N*J))/(I)
190 PRINT"LENING      = F1.";
200 PRINT USING"#####.##";P
210 LOCATE 0,21:PRINT"      NOG EEN BEREKEN
ING Ja/Nee ? "
220 Q$=INKEY$:IF Q$="" THEN 220
230 IF Q$="J" OR Q$="j" THEN RUN
240 IF Q$<>"N" AND Q$<>"n" THEN 220
250 END

```

Leningen 2: Berekenen van het aflossingsbedrag van een lening

Met dit programma berekent U welk bedrag U per aflossingsperiode moet betalen op een bepaalde lening.

De berekening geschiedt volgens de formule:

$$R = ((I/100) \times P/N) / (1 - 1/((I/100)/N + 1) \uparrow (N \times J))$$

waarbij: R = Aflossingsbedrag

P = Bedrag van lening

I = Jaarlijkse rentepercentage

N = Aantal aflossingen per jaar

J = Aantal jaren dat er afgelost moet worden

Voorbeeld:

Wat moet U betalen op een lening van f 12.000,00 tegen 8%, als U gedurende vijf jaar per maand moet terugbetalen?

REGELMATIGE AFLOSSINGEN.

XXXXXXXXXXXXXXXXXXXXXXXXXXXX

Aantal jaren : 5

Bedrag van lening : 12000

Jaarlijks rente pct. : 8

Aantal aflossingen(p/j): 12

XXXXXXXXXXXXXXXXXXXXXXXXXXXX

Aflossing = Fl 243.32

Nog een berekening ? (Y)ja of (N)ee)


```

10 REM *****
    **      LENINGEN  2      **
    *****
20 CLS:COLOR 15,1,1:WIDTH 40:KEYOFF
30 PRINT"          REGELMATIGE AFLOSSINGEN
":PRINT
40 PRINT"%%%%%%%%%" :PRINT
50 INPUT"AANTAL JAREN          ":"J
60 PRINT
70 INPUT"BEDRAG VAN LENING      ":"P
80 PRINT
90 INPUT"JAARLIJKS RENTEPCT.    ":"I
100 PRINT
110 INPUT"AANTAL AFLOSSINGEN P/J:";N
120 PRINT:PRINT"%%%%%%%%%"
    %%%%%%%%%%"
130 PRINT
150 REM *****
    **                                  **
    **      RENTE OMZETTEN VAN      **
    **      PROCENT NAAR DECIMAAL  **
    *****

160 I=I/100
170 REM *****
    **      BEREKEN NU HET          **
    **      TE LENEN BEDRAG         **
    **      D.M.V. DE FORMULE       **
    *****

180 R=(I*P/N)/(1-1/((I/N+1)^(N*J)))
190 PRINT"AFLOSSING          = F1.";
200 PRINT USING"#####.##";R
210 LOCATE 0,21:PRINT"      NOG EEN BEREKEN
ING Ja/Nee ? "
220 Q$=INKEY$:IF Q$="" THEN 220
230 IF Q$="J" OR Q$="j" THEN RUN
240 IF Q$<>"N" AND Q$<>"n" THEN 220
250 END

```

Leningen 3: Laatste betaling van een lening

Dit programma berekent voor U het laatste aflossingsbedrag van een lening of het bedrag dat er nog openstaat van een lening nadat U een X aantal malen heeft afgelost.

Het bedrag van de laatste aflossing is het bedrag van de regelmatige aflossing + de balans na N malen aflossen.

Voorbeeld:

Pieter heeft f 6000,00 geleend tegen 5%, als hij nu vijf jaar f 1000,00 aflost en hij besluit het restant ineens af te lossen, wat moet hij dan ineens betalen?

LAATSTE AFLOSSING VAN EEN LENING

XX

Bedrag van aflossing : 1000

Bedrag van lening : 6000

Aantal jaren : 5

Jaarlijks rente pct. : 5

Aantal aflossingen (p/j): 1

XX

Laatste aflossing = Fl 3132.06

Nog een berekening ? (Ja of Nee)

```

10 REM *****
    **   LAATSTE AFLOSSING   **
    **   VAN EEN LENING     **
    *****
20 CLS:WIDTH 40:COLOR 15,1,1:KEYOFF
30 PRINT"   LAATSTE AFLOSSING VAN EEN LE
NING"
40 PRINT:PRINT"%%%%%%%%%%%%%%
%%%%%%%%%%%%%":PRINT
50 INPUT"BEDRAG VAN AFLOSSING: ";R
60 PRINT
70 INPUT"BEDRAG VAN DE LENING: ";P
80 PRINT
90 INPUT"AANTAL JAREN           : ";J
100 PRINT
110 INPUT"JAARLIJKS RENTEPCT. : ";I
120 PRINT
130 REM *****
    **   VAN PROCENTEN     **
    **   NAAR DECIMALEN    **
    *****
140 I=I/100
150 INPUT"AANTAL AFLOSSINGEN(P/J):";N
160 B0=P
170 REM *****
    **BEREKEN ALLE BETALINGEN**
    ** EN MAAK DE BALANS OP  **
    *****
180 FOR Y= 1 TO N*J
190 REM *****
    ** AFRONDEN V/D BETAALDE **
    **       RENTE           **
    *****
200 U=INT((B0*I/N)*100+.5)/100
210 A=R-U
220 B0=B0-A
230 NEXT Y
240 REM *****
    ** BEREKEN DE LAATSTE   **
    ** AFLOSSING, DAARNA    **
    ** AFRONDEN EN PRINTEN. **
    *****

```

```

250 PRINT:PRINT"%%%%%%%%%%%%%%
%%%%%%%%%%%%%%":PRINT
260 PRINT"LAATSTE AFLOSSING =F1.";INT((R
+B0)*100+.5)/100
270 LOCATE 0,21:PRINT" NOG EEN BEREKENIN
G (Ja/Nee)"
280 Q$=INKEY$:IF Q$="" THEN 280
290 IF Q$="J" OR Q$="j" THEN RUN
300 IF Q$="N" OR Q$="n" THEN END
310 GOTO 280

```

Leningen 4: Berekenen van de looptijd van een lening

Dit programma berekent voor U de looptijd van een bepaalde lening. U moet daartoe de volgende gegevens invoeren!

- Het geleende bedrag
- Het aflossingsbedrag
- Het aantal aflossingen per jaar en
- De nominale rente

De berekening geschiedt dan met de formule:

$$J = -(\text{LOG}(1-(P \times I/100)) / (N \times R)) / ((\text{LOG}(1 + I/100/N) \times N))$$

waarbij: J = Aantal jaren dat de lening zal lopen

P = Het geleende bedrag

I = De nominale intrest

N = Aantal aflossingen per jaar

R = Bedrag van de periodieke aflossing

Voorbeeld:

De Familie Te Kort besluit f 15.000,00 te lenen tegen 11,5% rente. Ze betaalt per maand f 450,00 terug. Hoelang duurt de aflossing?

LOOPTIJD VAN EEN LENING

Bedrag van aflossing	: 450
Bedrag van lening	: 15000
Jaarlijks rente pct.	: 11.5
Aantal aflossingen (p/j)	: 12

Looptijd = 3.4 jaar

Nog een berekening ? (Ja of Nee)

```

10 REM *****
    **      LENINGEN      4      **
    *****
20 CLS:COLOR 15,1,1:WIDTH 40:KEYOFF
30 PRINT"      LOOPTIJD VAN EEN LENING":PR
INT
40 PRINT"%%%%%%%%%%%%%%"
"%%%%%%%%%":PRINT
50 INPUT"BEDRAG VAN AFLOSSING  ":";R
60 PRINT
70 INPUT"BEDRAG VAN LENING      ":";P
80 PRINT
90 INPUT"JAARLIJKS RENTEPCT.    ":";I
100 PRINT
110 INPUT"AANTAL AFLOSSINGEN P/J!":N
120 PRINT:PRINT"%%%%%%%%%%%%%%"
"%%%%%%%%%"
130 PRINT
150 REM *****
    **                                  **
    **      RENTE OMZETTEN VAN      **
    **      PROCENT NAAR DECIMAAL  **
    *****

160 I=I/100
170 REM *****
    **      BEREKEN NU DE          **
    **      LOOPTIJD              **
    **      D.M.V. DE FORMULE      **
    *****

175 J1=LOG(1-(P*I)/(N*R))
180 J2=LOG(1+I/N)*N
185 J=-(J1/J2)
190 PRINT"LOOPTIJD      =      ";
200 PRINT USING"##.## JAAR";J
210 LOCATE 0,21:PRINT"      NOG EEN BEREKEN
ING Ja/Nee ? "
220 Q$=INKEY$:IF Q$="" THEN 220
230 IF Q$="J" OR Q$="j" THEN RUN
240 IF Q$<>"N" AND Q$<>"n" THEN 220
250 END

```

Leningen 5: Berekenen van de nominale rente op een lening

Dit programma berekent voor U de nominale rente die U moet betalen op een lening. Tevens kunt U de vele aanbiedingen, die er voor leningen in de dagbladen verschijnen, ontrafelen.

U moet daartoe over de volgende gegevens beschikken:

- Het te lenen bedrag
- De totale looptijd
- Het aantal aflossingen per jaar
- Het periodieke aflossingsbedrag

De berekening vindt plaats met de zgn. benaderingsmethode. Om de werking van de methode te kunnen volgen zijn er extra REM-statements opgenomen.

Voorbeeld:

In de krant vonden we volgende aanbieding:

U kunt f 5.000,00 lenen en U betaalt terug: 60 maandelijkse termijnen van f 113,00.

Wat is nu het werkelijke rentepercentage, dat U moet betalen?

```
RENTÉ PCT. OP EEN LENING (P/J)
=====
Bedrag van lening           :5000
Aantal aflossingen (p/J):12
Bedrag van aflossing        :113
Aantal jaren                 :5
=====
Jaarlijks rente pct. = 12.699 %

Nog een berekening ? (Ja of Nee)
```

```

10 REM *****
    **      LENINGEN 5      **
    ** BEREKENEN NOMINALE **
    ** RENTE VAN 'N LENING **
    *****
20 CLS:WIDTH 40:COLOR 15,1,1:KEYOFF
30 PRINT"RENTE PCT. OP EEN LENING (P/J)"
40 PRINT:PRINT"%%%%%%%%%"
    %%%%%%%%%%":PRINT
50 INPUT "BEDRAG VAN LENING      ":";P
60 PRINT
70 INPUT "AANTAL AFLOSSINGEN(P/J):";N
80 PRINT
90 INPUT "BEDRAG VAN AFLOSSING  ":";R
100 PRINT
110 INPUT"AANTAL JAREN          ":";J
120 REM ** NEEM EEN RENTE VAN 10 % **
    ** OM TESTEN TE BEGINNEN **
130 I=10
140 REM ** BEREKEN DE REGELMATIGE **
    ** AFLOSSING MET GEBRUIK **
    ** VAN HET GEGOKTE PCT. **
150 R1=(I*P/N)/((1-1/((I/N+1)^(N*J)))
160 REM ** AFLOSSING AFRONDEN **
170 R1=INT(R1*100+.5)/100
180 REM ** I3 IS OM HET RENTEPCT. **
    ** AF TE SLUITEN. **
190 I3=ABS(I-I2)/2
200 REM ** BEWAAR DIT GETAL **
210 I2=I
220 REM ** VERGELIJK HET BEREKENDE **
    ** BEDRAG (R1) MET HET **
    ** GEGEVEN BEDRAG (R) **
230 REM *****
    ** ALS ZE GELIJK ZIJN, DAN **
    ** ZAL HET LAATSTE GEGOKTE **
    ** GETAL ONGEVEER HETZELFDE **
    ** ZIJN ALS DE REGELMATIGE **
    ** AFLOSSING. **
240 IF R1=R THEN 350

```



```

250 IF R1>R THEN 310
260 REM ** ALS R1<R DAN ZAL HET **
    ** RENTEPCT.HOGER ZIJN DAN **
    ** DE LAATSTE GOK. **
    *****
270 I=I+13
280 REM *****
    ** OPNIEUW TESTEN MET EEN **
    ** NIEUW GEGOKT GETAL **
    *****
290 GOTO 170
300 REM *****
    ** ALS R1>R DAN ZAL HET **
    ** RENTEPCT.LAGER ZIJN DAN **
    ** DE LAATSTE GOK **
    *****
310 I=I-13
320 REM *****
    ** OPNIEUW TESTEN MET EEN **
    ** NIEUW GEGOKT GETAL **
    *****
330 GOTO 170
340 REM *****
    ** BEREKEN DE RENTE NAAR **
    ** NORMALE GROOTTE,AFRONDEN**
    ** EN PRINTEN **
    *****
350 I=((INT((I*1000)*100+.5))/100)/1000
360 PRINT:PRINT"%%%%%%%%%"
    "%%%%%%%%%";PRINT
370 PRINT"JAARLIJKS RENTEPCT.= ";I*100;"
    % "
380 LOCATE 0,21:PRINT" NOG EEN BEREKENIN
    G (Ja/Nee)"
390 Q$=INKEY$:IF Q$="" THEN 390
400 IF Q$="J" OR Q$="j" THEN RUN
410 IF Q$="N" OR Q$="n" THEN END
420 GOTO 390

```

Leningen 6: Berekenen van het openstaand bedrag van een lening

Dit programma berekent voor U het bedrag, dat nog openstaat van een lening, nadat U reeds meerdere malen heeft afgelost.

Het bedrag dat berekend wordt, is het aflossingsbedrag, dat U moet betalen als U de lening in een keer wilt aflossen op het moment van berekening.

Wacht U een aflossingstermijn, dan komt er uiteraard weer rente bij.

Voorbeeld:

U heeft f 8000,00 geleend tegen 17,2% rente. U lost maandelijks f 200,00 af. Wat staat er open van deze lening, nadat U reeds 4 jaar en 10 maanden heeft afgelost?

OPENSTAAND BEDRAG VAN EEN LENING

Bedrag van aflossing : 200

Bedrag van lening : 8000

Aantal aflossingen (p/j): 12

Jaarlijks rente pct. : 17.2

In welk jaar was
de laatste aflossing : 4

De hoeveelste aflossing
was dat in dat jaar : 10

Openstaand bedrag = f 2496.17

```

10 REM *****
    ** OPENSTAAND BEDRAG **
    ** VAN EEN LENING **
    *****
20 CLS:WIDTH 40:COLOR 15,1,1:KEYOFF
30 PRINT " OPENSTAAND BEDRAG VAN EEN LE
NING"
40 PRINT:PRINT"%%%%%%%%%%%%%%
%%%%%%%%%%%%%%":PRINT
50 INPUT"BEDRAG VAN AFLOSSING: ";R
60 PRINT
70 INPUT"BEDRAG VAN DE LENING: ";P
80 PRINT
90 INPUT"AANTAL AFLOSSINGEN(P/J):";N
100 PRINT
110 INPUT"JAARLIJKS RENTEPCT. : ";I
120 PRINT
130 REM *****
    ** VAN PROCENTEN **
    ** NAAR DECIMALEN **
    *****
140 I=I/100
150 INPUT"IN WELK JAAR WAS
DE LAATSTE AFLOSSING :";J
160 PRINT
170 INPUT"DE HOVEELSTE AFLOSSING
WAS DAT IN DAT JAAR :";N1
180 B0=P
190 REM *****
    **BEREKEN ALLE BETALINGEN**
    ** EN MAAK DE BALANS OP **
    *****
200 FOR Y= 1 TO N*(J-1)+N1
210 REM *****
    ** BEREKEN DE RENTE PER **
    ** AFLOSSING **
    *****

```

```

220 I1=INT((B0*I/N)*100+.5)/100
230 A=R-I1
240 B0=B0-A
250 NEXT Y
260 REM *****
    ** BEREKEN OPENSTAAND **
    ** BEDRAG , DAARNA **
    ** AFRONDEN EN PRINTEN. **
    *****
270 PRINT:PRINT"%%%%%%%%%"
"%%%%%%%%%":PRINT
280 PRINT"OPENSTAAND BEDRAG =F1.";INT(B0
*100+.5)/100
290 LOCATE 0,21:PRINT" NOG EEN BEREKENIN
G (Ja/Nee)"
300 Q$=INKEY$:IF Q$="" THEN 300
310 IF Q$="J" OR Q$="j" THEN RUN
320 IF Q$="N" OR Q$="n" THEN END
330 GOTO 300

```

HYPOTHEEK- AFLOSSINGSTABEL

Met dit programma krijgt U van een hypotheek een compleet overzicht in tabelvorm over de betaalde rente per aflossingsperiode alsmede het nog openstaande bedrag. Daarnaast wordt per jaar een overzicht gegeven van de in dat jaar betaalde rente en werkelijke aflossing.

Voorbeeld:

U heeft een aardig optrekje op het oog. U had al zo lang zin in een eigen dak boven Uw hoofd en U besluit het na rijp beraad met Uw vrouw toch maar te kopen, want de rente is nog steeds aftrekbaar en daarmee rekening houdend is dit voor U een haalbare kaart.

U had ook nog een spaarcentje achter de hand en U moet daarnaast een HYPOTHEEK afsluiten van f 135.000,00 tegen 8,75% (met gemeente garantie). De looptijd is vastgesteld op 30 jaar. Hoe ziet de aflossingstabel er uit?

Bedrag van lening		:	FL 135000
Rente		:	8,75 %
Looptijd in jaren		:	30
Aflossingen		:	FL 1062.05

Nr.	Rente	Afl.	Salans
1	984.38	77.67	134922.33
2	983.81	78.24	134844.09
3	983.24	78.81	134765.28
4	982.65	79.39	134685.89
5	982.08	79.97	134605.92
6	981.51	80.55	134525.37
7	980.91	81.14	134444.23
8	980.32	81.73	134362.5
9	979.73	82.32	134280.18
10	979.13	82.92	134197.26
11	978.52	83.53	134113.73
12	977.91	84.14	134029.59

Maat	Rente	Afgelost
1	11774.19	970.41

Bedrag van lening : FL 135000
 Rente : 8.75 %
 Looptijd in jaren : 30
 Aflossingen : FL 1062.05

Nr.	Rente	Afl.	Salans
1	312.64	749.41	42127.43
2	307.18	754.87	41372.56
3	301.67	760.38	40612.18
4	296.13	765.93	39846.25
5	290.55	771.50	39074.76
6	284.93	777.13	38297.63
7	279.28	782.80	37514.83
8	273.59	788.50	36726.33
9	267.86	794.25	35932.08
10	262.09	800.05	35132.03
11	256.17	805.89	34326.16
12	250.29	811.76	33514.39

Jaar	Rente	Afgelost
27	3382.15	9362.45

Bedrag van lening : FL 135000
 Rente : 8.75 %
 Looptijd in jaren : 30
 Aflossingen : FL 1062.05

Nr.	Rente	Afl.	Salans
1	88.62	973.43	11179.73
2	81.52	980.53	10199.2
3	74.37	987.68	9211.52
4	67.17	994.88	8216.64
5	59.91	1002.14	7214.5
6	52.61	1009.44	6205.06
7	45.25	1016.8	5188.26
8	37.83	1024.22	4164.04
9	30.36	1031.69	3132.35
10	22.84	1039.21	2093.14
11	15.26	1046.79	1046.35
12	7.63	1046.35	0

Laatste aflossing	: FL 1053.98	
Jaar	Rente	Afgelost
30	583.37	12153.16

```

10 REM *****
    **                                     **
    **          HYPOTHEEK-              **
    **    AFLOSSINGSTABEL              **
    **                                     **
    *****
20 CLS:WIDTH 40:COLOR 15,1,1:KEYOFF
30 PRINT"          HYPOTHEEK-AFLOSSINGSTABEL":
PRINT
40 PRINT"%%%%%%%%%%%%%%"
"%%%%%%%%%":PRINT
50 INPUT"BEDRAG REGELM. AFLOSSING: ";R
60 PRINT
70 INPUT"LOOPTIJD IN JAREN          ";J
80 PRINT
90 INPUT"BEDRAG VAN HYPOTHEEK      ";P
100 PRINT
110 INPUT"JAARLIJKS RENTEPCT.      ";I
120 PRINT
130 REM *****
    ** VAN PROCENT NAAR DEC.**
    *****
140 I=I/100
150 INPUT"AANTAL AFLOSSINGEN (P/J): ";N
160 PRINT
170 INPUT"VANAF WELK JAAR PRINTEN ";X
180 PRINT
190 REM *****
    ** PRINTEN VANAF BEGIN **
    **    VAN EEN JAAR    **
    *****
200 X=INT(X)
210 REM *****
    **                                     **
    **    INITIALISEREN VAN          **
    **    HULPVARIABLEN              **
    **                                     **
    *****
220 C1=0:I2=0:I3=0
230 N1=N:K=0:Q=5

```

```

240 B0=P:A1=0:A2=0
250 REM *****
    ** LOOPTIJD MINDER DAN **
    **      1 JAAR ?      **
    *****
260 IF INT(J)<1 THEN 920
270 REM *****
    ** EEN FOR-NEXT LOOP **
    **   VOOR ELK JAAR   **
    *****
280 FOR Y=1 TO INT(J)
290 REM *****
    ** AL BEGINNEN MET HET **
    **      PRINTEN ?      **
    *****
300 IF J<X THEN 410
310 IF K=1 THEN 410
320 K=1
330 CLS
340 REM *****
    ** PRINTEN VAN DE KOP **
    *****
350 PRINT"HYPOTHEEKBEDRAG      ":";USING"F1
    #####.##";P
360 PRINT"RENTE                  ":";I*100;"
    %"
370 PRINT"LOOPTIJD IN JAREN    ":";J
380 PRINT"AFLOSSINGEN          ":";USING"F1
    ####.##";R
390 PRINT"%%%%%%%%%%%%%%"
    %%%%%%%%%%"
400 PRINT" NO:   RENTE      AFLOSS.      BALA
NS"
410 FOR U=1 TO N1
420 REM *****
    **   BEREKEN DE BETAALDE **
    **   RENTE OVER DEZE   **
    ** AFLOSSING EN PRINTEN **
    *****
430 I1=INT((B0*I/N)*100+.5)/100

```



```

440 REM *****
      ** TEL HET AANTAL TOT NU **
      ** TOE GEDANE AFLOSSINGEN **
      *****
450 C1=C1+1
460 REM *****
      ** BEREKEN HET BEDRAG DAT **
      ** IS AFGELOST IN DEZE **
      ** PERIODE **
      *****
470 A=R-I1
480 REM *****
      ** DE AFGELOSTE BEDRAGEN **
      ** OPTELLEN **
      *****
490 A1=A1+A
500 REM *****
      ** BEREKEN NU DE BALANS **
      *****
510 B0=P-A1
520 REM *****
      ** DE LAATSTE AFLOSSING ? **
      ** ZO JA,BEREKEN DEZE DAN **
      *****
530 IF C1<>N*J THEN 590
540 R=R+B0
550 A=A+B0
560 A1=A1+B0
570 B0=0
580 REM *****
      ** TEL DE TOT NU TOE **
      ** BETAALDE RENTE OP **
      *****
590 I2=I2+I1
600 REM *****
      ** TEL DE IN DIT JAAR **
      ** BETAALDE RENTE OP **
      *****
610 I3=I3+I1

```

```

620 REM *****
    ** TEL HET IN DIT JAAR **
    ** AFGELOSTE BEDRAG OP **
    *****
630 A2=A2+A
640 REM *****
    ** BEGINNEN MET PRINTEN ? **
    ** ZO JA, ZET DAN DE **
    ** BEREKENDE WAARDEN IN **
    ** EEN TABEL. **
    *****
650 IF Y<X THEN 680
660 Q=Q+1:IF Q>21 THEN GOSUB 1030
670 PRINT U;:PRINT TAB(4);USING"####.##"
;I1;:PRINT TAB(15);USING"####.##";INT(A*
100+.5)/100;:PRINT TAB(25);USING"#####.
##";INT(B0*100+.5)/100
680 NEXT U
690 REM *****
    ** LAATSTE AFLOSSING ? **
    ** ZO JA, DAN AFRONDEN **
    ** EN PRINTEN **
    *****
700 IF C1<>N*J THEN 750
710 Q=Q+4:IF Q>21 THEN GOSUB 1030
720 PRINT
730 PRINT"LAATSTE AFLOSSING F1: ";USING
"####.##";INT(R*100+.5)/100
740 GOTO 950
750 IF Y<X THEN 840
760 Q=Q+4:IF Q>21 THEN GOSUB 1030
770 PRINT
780 PRINT" JAAR RENTE AFGELOST"
790 GOTO 810
800 IF Y>INT(J+1) THEN Y=INT(J+1):K=1
810 PRINT Y;:PRINT TAB(10);USING"####.##"
;I3;:PRINT TAB(20);USING"#####.##";A2
820 PRINT
830 REM *****
    ** LOOPTIJD AFGELOPEN ? **
    *****

```

```

840 IF Y>J THEN 950
850 REM *****
      * EEN AANTAL HULPVARIABLEN*
      *   OPNIEUW INSTELLEN   *
      * VOOR HET VOLGENDE JAAR *
      *****

860 I3=0
870 A2=0
880 NEXT Y
890 REM*****
      ** PRINTEN VAN EEN BEPAALD**
      **           JAAR ?           **
      *****

900 IF J=Y THEN 950
910 REM*****
      **   BIJSTELLEN VAN DE   **
      **   VARIABLEN OM EEN    **
      ** BEPAALD JAAR TE PRINTEN**
      *****

920 N1=((J-INT(J))*12)/12*N
930 Y=Y+1
940 GOTO 300
950 PRINT" JAAR           RENTE           AFGELOST"
960 IF Y>INT(J+1) THEN Y=INT(J+1)
970 PRINT Y;:PRINT TAB(10);USING"#####.#
#";I3;:PRINT TAB(20);USING"#####.##";A2
980 LOCATE 0,22:PRINT" NOG EEN BEREKENIN
G ? (Ja/Nee)"
990 Q$=INKEY$:IF Q$="" THEN 990
1000 IF Q$="J" OR Q$="j" THEN RUN
1010 IF Q$="N" OR Q$="n" THEN END
1020 GOTO 990
1030 LOCATE 0,22:PRINT" TOETS >RETURN< V
OOR VOLGENDE PAGINA"
1040 Q$=INKEY$:IF Q$="" THEN 1040
1050 IF ASC(Q$)<>13 THEN 1040
1060 Q=6
1070 CLS
1080 PRINT"HYPOTHEEKBEDRAG           :";USING"F
l #####.##";P

```

```

1090 PRINT"RENTE                :";I*100;"
%"
1100 PRINT"LOOPTIJD IN JAREN  :";J
1110 PRINT"AFLOSSINGEN        :";USING"F
1  ####.##";R
1120 PRINT"%%%%%%%%%%"
%%%%%%%%%%"
1130 PRINT" NO:  RENTE      AFLOSS.      BAL
ANS"
1140 RETURN

```

AFSCHRIJVEN

Afschrijvingspercentage

Met dit programma berekent U precies het afschrijvingspercentage van een bepaalde aankoop.

De formule voor de berekening is: $A = 100 \times (1 - (T/P) \div (1/J))$

waarbij: A = Afschrijvingspercentage

P = Aankoopsprijs

T = Restwaarde na J jaren gebruikt

J = Aantal jaren in gebruik

Voorbeeld:

Constructiebedrijf JANSEN BV heeft in 1972 een puntlasmachine gekocht voor f 20.000,00 en de machine na 12 jaar gebruik verkocht voor f 1.850,00. Wat is het afschrijvingspercentage geweest voor deze machine?

AFSCHRIJVINGS-PERCENTAGE

Aankoops prijs	: 20000
Verkoops prijs	: 1850
Aantal jaren	: 12

Afschrijvings pct. = 17.994 %

Nog een berekening ? (Ja of Nee)

```

10 REM *****
**      AFSCHRIJVEN 1      **
**                               **
** HET AFSCHRIJVINGSPCT**
*****
20 CLS:COLOR 15,1,1:WIDTH 40:KEYOFF
30 PRINT"      AFSCHRIJVINGS-PERCENTAGE"
:PRINT
40 PRINT"%%%%%%%%%%%%%%%"
"%%%%%%%%%%":PRINT
50 INPUT"AANKOOPSPRIJS      ":";P
60 PRINT
70 INPUT"VERKOOPSPRIJS      ":";T
80 PRINT
90 INPUT"AANTAL JAREN GEBRUIKT ":";J
100 PRINT:PRINT"%%%%%%%%%%%%%%%"
"%%%%%%%%%%%"
110 PRINT
120 REM *****
**                               **
**      BEREKEN NU HET      **
**AFSCHRIJVINGS-PERCENTAGE**
**      D.M.V.  DE FORMULE  **
*****
130 A=100*(1-(T/P)^(1/J))
140 PRINT"AFSCHRIJVINGSPCT.= ";
150 PRINT USING"###.## %";A
160 LOCATE 0,21:PRINT"      NOG EEN BEREKEN
ING Ja/Nee ? "
170 Q$=INKEY$:IF Q$="" THEN 170
180 IF Q$="J" OR Q$="}" THEN RUN
190 IF Q$<>"N" AND Q$<>"n" THEN 170
200 END

```

Afschrijvingsbedrag

In tegenstelling tot het voorafgaande programma berekent dit programma het bedrag dat jaarlijks wordt afgeschreven op een aankoop.

De formule hiervoor is: $A = P \times I \times (1 - I)^{J-1}$

waarbij: A = Afschrijvingsbedrag

P = Aankoopbedrag

I = Afschrijvingspercentage

J = Aantal jaren van afschrijving

Voorbeeld:

Jacques kocht een auto van f 18.890,00. Hij schrijft per jaar 21% af. Wat zijn de afschrijvingsbedragen per jaar gedurende de eerste drie jaren?

AFSCHRIJVINGS-BEDRAG

Aankoops prijs : 18890

Afschrijvings-percentage: 21

Welk jaar : 1

Afschrijving = FL 3966.9

Welk ander jaar? (0=STOPPEN)

AFSCHRIJVINGS-BEDRAG

Rankoops prijs : 18890

Afschrijvings-percentage: 21

Welk jaar : 2

Afschrijving = Fl 3133.85

Welk ander jaar? (@=STOPPEN)

AFSCHRIJVINGS-BEDRAG

Rankoops prijs : 18890

Afschrijvings-percentage: 21

Welk jaar : 3

Afschrijving = Fl 2475.74

Welk ander jaar? (@=STOPPEN)


```

10 REM *****
    **      AFSCHRIJVEN 2      **
    **                               **
    ** AFSCHRIJVINGSBEDRAG **
    *****
20 CLS:COLOR 15,1,1:WIDTH 40:KEYOFF
30 PRINT"      AFSCHRIJVINGS-BEDRAG":PRI
NT
40 PRINT"%%%%%%%%%"
"%%%%%%%%%":PRINT
50 INPUT"AANKOOPSPRIJS      ":";P
60 PRINT
70 INPUT"AFSCHRIJVINGSPCT.      ":";I
80 REM *****
    ** PERCENTAGE OMZETTEN VAN **
    ** PROCENT NAAR DECIMAAL **
    *****
90 I=I/100
100 PRINT
110 INPUT"WELK JAAR      ":";J
120 IF J=0 THEN 210
130 PRINT:PRINT"%%%%%%%%%"
"%%%%%%%%%"
140 PRINT
150 REM *****
    **                               **
    **      BEREKEN NU HET      **
    ** AFSCHRIJVINGS-BEDRAG **
    ** D.M.V. DE FORMULE **
    *****
160 A=P*I*(1-I)^(J-1)
170 PRINT"AFSCHRIJVING      = ";
180 PRINT USING"F1. #####.## ";A
190 LOCATE 0,21:PRINT"      WELK ANDER JAA
R (0= STOPPEN)      "
200 LOCATE 0,9:GOTO 110
210 LOCATE 0,21:PRINT"      NOG EEN BEREK
ENING (Ja/Nee)      "
220 Q$=INKEY$:IF Q$="" THEN 210
230 IF Q$="J" OR Q$="}" THEN RUN
240 IF Q$<>"N" AND Q$<>"n" THEN 210
250 END

```

Restwaarde

Dit programma berekent de restwaarde van een produkt nadat gedurende een aantal jaren een bepaald percentage is afgeschreven.

De berekening wordt gedaan met de formule: $R = (P \times (1-I))^J$

waarbij: R = Restwaarde

P = Aanschafsprijs

I = Afschrijvingspercentage

J = Aantal jaren dat er afgeschreven wordt

Voorbeeld:

Jo-Anne kocht een HiFi-installatie voor f 950,00. Wat is de waarde van de installatie na 2 jaar bij een afschrijvingspercentage van 25%?

RESTWAARDE

Aankoops prijs : 950
Afschrijvings pct. : 25
Aantal jaren : 2

Restwaarde = FL 534.38

Ander aantal jaren? (@=STOPPEN)

```

10 REM *****
    **      AFSCHRIJVEN 3      **
    **                                     **
    **      RESTWAARDE      **
    *****
20 CLS:COLOR 15,1,1:WIDTH 40:KEYOFF
30 PRINT"      RESTWAARDE BEREKENEN":PRI
NT
40 PRINT"%%%%%%%%%"
"%%%%%%%%%":PRINT
50 INPUT"AANKOOPSPRIJS      ":";P
60 PRINT
70 INPUT"AFSCHRIJVINGSPCT.      ":";I
80 REM *****
    ** PERCENTAGE OMZETTEN VAN **
    ** PROCENT NAAR DECIMAAL **
    *****
90 I=I/100
100 PRINT
110 INPUT"AANTAL JAREN      ":";J
120 IF J=0 THEN 210
130 PRINT:PRINT"%%%%%%%%%"
"%%%%%%%%%"
140 PRINT
150 REM *****
    **                                     **
    **      BEREKEN NU DE      **
    **      RESTWAARDE      **
    **      D.M.V. DE FORMULE      **
    *****
160 R=P*(1-I)^J
170 PRINT"RESTWAARDE      =" ";
180 PRINT USING"F1. #####.## ";R
190 LOCATE 0,21:PRINT"      WELK ANDER JAA
R (0= STOPPEN)      "
200 LOCATE 0,9:GOTO 110
210 LOCATE 0,21:PRINT"      NOG EEN BEREK
ENING (Ja/Nee)      "
220 Q$=INKEY$:IF Q$="" THEN 210
230 IF Q$="J" OR Q$="j" THEN RUN
240 IF Q$<>"N" AND Q$<>"n" THEN 210
250 END

```

ECONOMISCHE BESTELHOEVEELHEID

Dit programma berekent voor U welk aantal van een bepaald produkt U moet bestellen om zo economisch mogelijk te werken.

De berekening vindt plaats op basis van de jaarlijkse behoefte, de stuksprijs, de bestel- en opslagkosten. De computer geeft aan welke hoeveelheid U moet bestellen en geeft daarna een overzicht van diverse andere bestelmogelijkheden met een opsomming van de kosten voor bestellen en opslaan per bepaalde hoeveelheid.

Is het niet mogelijk een bepaalde economische besteleenheid te berekenen dan wordt dit door de computer ook aangegeven.

Voorbeeld:

U heeft per jaar van een bepaald produkt 250 stuks nodig. U kunt die inkopen tegen een stuksprijs van f 1,45. Het opslaan van dit produkt in het magazijn kost U f 0,45 op jaarbasis. Bovendien betaald U per bestelling die U doet f 22,50 bestel- en administratiekosten. Wat is nu het meest economische aantal dat U per keer moet bestellen?

```
BEREKENING ECONOMISCHE
BESTELHOEVEELHEID

GEEF VOLGENDE GEGEVENS:

STUKSPRIJS: 1.45
BENODIGD AANTAL/JAAR: 250
OPSLAGKOSTEN/STUK/JAAR: 0.45
KOSTEN PER ORDER: 22.5
XXXXXXXXXXXXXXXXXXXXXXXXXXXX
U MOET 158 STUKS BESTELLEN OM
ECONOMISCH VERANTWOORD TE ZIJN.

TOETS ENTER VOOR VERVOLG
```

VERDERE GEGEVENS VAN DIT PRODUCT

OM 104 STUKS TE BESTELLEN
ZIJN DE KOSTEN VOOR OPSLAG EN
BESTELLING FL.23.62

VERDERE GEGEVENS VAN DIT PRODUCT

OM 156 STUKS TE BESTELLEN
ZIJN DE KOSTEN VOOR OPSLAG EN
BESTELLING FL.35.39

VERDERE GEGEVENS VAN DIT PRODUCT

OM 208 STUKS TE BESTELLEN
ZIJN DE KOSTEN VOOR OPSLAG EN
BESTELLING FL.47.12

VERDERE GEGEVENS VAN DIT PRODUCT

OM 260 STUKS TE BESTELLEN
ZIJN DE KOSTEN VOOR OPSLAG EN
BESTELLING FL.58.85

VERDERE GEGEVENS VAN DIT PRODUCT

OM 312 STUKS TE BESTELLEN
ZIJN DE KOSTEN VOOR OPSLAG EN
BESTELLING FL.70.56

VERDERE GEGEVENS VAN DIT PRODUCT

OM 364 STUKS TE BESTELLEN
ZIJN DE KOSTEN VOOR OPSLAG EN
BESTELLING FL.82.27

VERDERE GEGEVENS VAN DIT PRODUCT

OM 416 STUKS TE BESTELLEN
ZIJN DE KOSTEN VOOR OPSLAG EN
BESTELLING FL.93.98

NOG EEN BEREKENING (JA/NEE)

```

10 REM *****
    **      ECONOMISCHE      **
    **      BESTELHOEVEELHEID  **
    *****
20 CLS:WIDTH 40:COLOR 15,1:KEYOFF
30 PRINT"          BEREKENING ECONOMISCHE
          BESTELHOEVEELHEID"
40 PRINT:PRINT"%%%%%%%%%"
%%%%%%%%%" :PRINT
50 PRINT" GEEF DE VOLGENDE GEGEVENS:" :PR
INT
60 INPUT"      INKOOPSPRIJS PER STUK: ";S
70 PRINT
80 INPUT"      BENODIGD AANTAL/JAAR: ";N
90 PRINT
100 INPUT"     OPSLAGKOSTEN/STUK/JAAR: ";O
110 PRINT
120 INPUT"                      BESTELKOSTEN: ";B
130 PRINT:PRINT"%%%%%%%%%"
%%%%%%%%%" :PRINT
140 X=(2*N*B)/O
150 GOSUB 380
160 H=INT(Y)
170 IF H=0 THEN 450
180 PRINT"U MOET ";H;" STUKS BESTELLEN O
M":PRINT"ECONOMISCH VERANTWOORD TE BESTE
LLEN."
190 LOCATE 0,21:PRINT"  TOETS >RETURN<
VOOR VERVOLG"
200 Q$=INKEY$:IF Q$="" THEN 200
210 IF ASC(Q$)<>13 THEN 200
220 CLS
230 H=H/3:H=INT(H)
240 R=H:N=0
250 N=N+1
260 H=H+R
270 T=N/H*B+H/2*O
280 T=INT(T*100+.5)/100
290 IF N>7 THEN 460
300 CLS:PRINT"VERDERE GEGEVENS VAN DIT P
RODUKT:"

```

```

310 PRINT:PRINT"OM ";H;" STUKS TE BESTEL
LEN"
320 PRINT"ZIJN DE KOSTEN VOOR OPSLAG EN
      BESTELLING: ";USING"#####.##";
T
330 LOCATE 0,22:PRINT" TOETS >RETURN< VO
OR VERVOLG "
340 Q$=INKEY$:IF Q$="" THEN 340
350 IF ASC(Q$)<>13 THEN 340
360 LOCATE 0,22:PRINT SPACE$(40)
370 GOTO 250
380 IF X=0 THEN Y=0:RETURN
390 IF X>0 THEN 410
400 PRINT:PRINT"U MAAKTE EEN FOUT BIJ HE
T INVOEREN !!":FOR V=1 TO 1500:NEXT V:GO
TO 10
410 Y=INT(X*.5):Z=0
420 W=INT((X/Y-Y)*.5)
430 IF W=0 THEN RETURN
440 Y=Y+W:Z=W:GOTO 420
450 LOCATE 0,15:PRINT"DIT PRODUKT KUNT U
NIET IN EEN BEPAALD ECONOMISCH AANTAL
BESTELLEN"
460 CLS:LOCATE 0,21:PRINT" NOG EEN BEREK
ENING ? (Ja/Nee)"
470 Q$=INKEY$:IF Q$="" THEN 470
480 IF Q$="J" OR Q$="j" THEN RUN
490 IF Q$="N" OR Q$="n" THEN END
500 GOTO 470

```

BREAK-EVEN POINT ANALYSE

Met dit programma berekent U bij welk verkocht aantal van een bepaalde produktie U de kostprijs van de totale produktie heeft terug verdiend. Dit heet het BREAK-EVENPOINT. Alles wat U meer verkoopt dan dit BREAK-EVENPOINT levert U winst op. Een ideaal programma voor hen die met bepaalde produktieaantallen moeten werken en graag van te voren willen weten bij welk verkocht aantal de produktiekosten al zijn terugverdiend.

Voorbeeld:

Fietsenfabriek "De Gouden Doortrapper" besluit om een nieuw model aan haar assortiment toe te voegen. Dit model "SPORT 2000" wil men in een eerste produktierun van 10.000 stuks maken. Men heeft een reclamecampagne voor dit nieuwe model begroot van f 80.000,00 en daarnaast is nog een bedrag van f 20.000,00 nodig voor verpakking. De totale overhead is f 100.000,00. De directe produktiekosten zijn inclusief materialen en arbeidsloon etc. totaal: f 1.500.000,00. De verkoops prijs aan de rijwielhandel is: f 295,00. Bij welk aantal verkochte rijwielen is men uit de aanloopkosten van dit model?

BREAK EVEN POINT ANALYSE

GEEF DE VOLGENDE GEGEVENS:

VARSTE KOSTEN(OVERHEAD): 100000

AANTAL TE PRODUCEREN: 10000

DIRECTE PRODUCTIEKOSTEN: 1500000

VERKOOPSPRIJS/STUK: 295

=====

BREAK-EVENPOINT LIGT OP: 5424
STUKS VAN DE 10000 TOTAAL.

BREAK-EVEN BIJ: 54.24 %

NOG EEN ANALYSE JA/NEE ??


```

10 REM *****
    **
    **   BEREKENEN   VAN HET   **
    **   BREAK-EVEN POINT   **
    **
    *****
20 CLS:COLOR 15,1,1:WIDTH 40:KEYOFF
30 PRINT"          BREAK-EVEN POINT ANALYSE
  ":PRINT
40 PRINT"%%%%%%%%%%%%%%
%%%%%%%%%"
50 PRINT">>          VERSTREK DE VOLGENDE GEGE
VENS <<"
60 PRINT
70 INPUT" VASTE KOSTEN (OVERHEAD) :";K
80 PRINT
90 INPUT"          AANTAL TE PRODUOEREN :";N
100 PRINT
110 INPUT" DIRECTE PRODUCTIEKOSTEN :";P
120 PRINT
130 INPUT"          VERKOOPSPRIJS/STUK :";V
140 PRINT
150 PRINT"%%%%%%%%%%%%%%
%%%%%%%%%":PRINT
160 REM *****
    ** KOSTPRIJS PER EENHEID **
    *****
170 E=INT(((K+P)/N)*100+.5)/100
180 W=V-E
190 S=INT((K+P)/V+.5)
200 PRINT"BREAK-EVEN POINT LIGT OP:";S
210 PRINT"STUKS VAN DE ";N;" TOTAAL."
220 PRINT:PRINT"BREAK-EVEN BIJ: ";(S/N)*
100;" %"
230 LOCATE 0,21:PRINT"          NOG EEN ANA
LYSE   (Ja/Nee) "
240 Q$=INKEY$:IF Q$="" THEN 240
250 IF Q$="J" OR Q$="j" THEN RUN
260 IF Q$<>"N" AND Q$<>"h" THEN 240
270 END

```

EENHEIDSPRIJS- BEREKENING

Met dit programma kunt exact uitrekenen welk de beste koop voor U zal zijn. Dit programma rekent namelijk de prijs per eenheid uit; dit kan zijn volume, gewicht of aantal.

Voorbeeld:

U krijgt een aanbieding van merk X. Zij leveren U het produkt in verpakkingen van 500 gram tegen een prijs van f 12,58 per 500 gr. Merk Y biedt U hetzelfde product aan in verpakkingen van 750 gram tegen een prijs van f 18,68. Wat is voor U nu de beste koop?

Met dit programma kunt U dat direct te weten komen. Dit voorbeeld is nog simpel en zelfs nog snel met de hand uit te rekenen, maar bij wat meer ingewikkelde verhoudingen is het voor U een voordeel dit programma bij de hand te hebben.

EENHEIDSPRIJS	
VERSTREK VOLGENDE GEGEVENS:	
NAAM PRODUKT 1	: X
PRIJS PRODUKT 1	: 12.58
VOLUME PRODUKT 1	: 500
NAAM PRODUKT 2	: Y
PRIJS PRODUKT 2	: 18.68
VOLUME PRODUKT 2	: 750
X	2.52 CENT/EENH.
Y	2.49 CENT/EENH.
Y	IS DE BESTE KEUS.
NOG EEN BEREKENING JA/NEE ?	

```

10 REM *****
    **
    **      BEREKENEN VAN      **
    **      EENHEIDSPRIJZEN    **
    **
    *****
20 CLS:COLOR 15,1,1:WIDTH 40:KEYOFF
30 PRINT "      EENHEIDSPRIJS ":PRINT
40 PRINT "%%%%%%%%%%%%%%"
    %%%%%%%%%%"
50 PRINT">>      VERSTREK DE VOLGENDE GEGE
VENS <<"
60 PRINT
70 INPUT"      NAAM PRODUKT 1 :";A$
80 INPUT"      PRIJS PRODUKT 1 :";P1
90 INPUT"      VOLUME PRODUKT 1 :";V1
100 PRINT
110 INPUT"      NAAM PRODUKT 2 :";B$
120 INPUT"      PRIJS PRODUKT 2 :";P2
130 INPUT"      VOLUME PRODUKT 2 :";V2
140 PRINT
150 PRINT"%%%%%%%%%%%%%%"
    %%%%%%%%%%" :PRINT
160 REM *****
    ** KOSTPRIJS PER EENHEID **
    *****
170 REM *****
    **
    **      LET ER OP DAT U BIJ      **
    **      INVOEREN VAN DE          **
    **      VOLUME-EENHEDEN          **
    **
180 REM **
    ** GELIJKE SOORTEN EENHEDEN**
    ** GEBRUIKT, VERGELIJK      **
    ** GEEN KILO'S MET LITERS!    **
    *****
190 E1=(P1/V1)*100:E2=(P2/V2)*100
200 PRINT A$,:PRINT USING"F1 ####.## /
    EENHEID";E1
210 PRINT B$,:PRINT USING"F1 ####.## /
    EENHEID";E2

```

```

220 IF E1=E2 THEN LOCATE 0,18:PRINT "BEI
DE PRODUKTEN ZIJN GELIJK GEPRIJSD "
230 IF E2<E1 THEN LOCATE 0,18:PRINT B$;"
  IS DE BESTE KEUS  ":GOTO 250
240 IF Q$="J" OR Q$="j" THEN RUN
250 LOCATE 0,21:PRINT"  NOG EEN BEREKENI
NG (Ja/Nee)"
260 Q$=INKEY$:IF Q$="" THEN 260
270 IF Q$="J" OR Q$="j" THEN RUN
280 IF Q$<>"N" AND Q$<>"n" THEN 260
290 END

```

VASTE LASTEN BEWAKING

Met dit programma is het nu mogelijk om per maand een gedetailleerd overzicht te maken van Uw vaste lasten. Daarnaast wordt alles gecumuleerd in een jaaroverzicht.

Het programma wordt volledig MENU-gestuurd.

U dient eerst gegevens in te voeren, zodat deze later op cassette kunnen worden gezet.

In totaal kunt U 50 rubrieken "vaste lasten" benoemen. Veranderingen kunt U verwerken door eerst de niet (meer) gewenste rubriek af te voeren en daarna weer nieuwe gegevens in te voeren.

Per rubriek kunt U de gegevens verder specificeren met behulp van frequentie (aantal malen per jaar dat er betaald moet worden), in welke maanden er betaald dient te worden en of er nog een jaarafrekening komt (zoals bijv. bij gas, water en electriciteit).

In het hierna gegeven voorbeeld is een vaste-lasten-bewaking gegeven van een huisgezin.

Maar U kunt het programma even zo goed gebruiken voor zakelijke toepassingen.

J A A R O V E R Z I C H T .

Huur.....	FL	10332.00
Gas.....	FL	3216.00
Televisiegids.....	FL	47.75
Electriciteit.....	FL	1404.00
Kabeltelevisie.....	FL	170.00
Telefoon.....	FL	1080.00
Dagblad.....	FL	254.44
Ziekenfonds.....	FL	60.00
Omroepbijdrage.....	FL	150.00
Weekblad.....	FL	35.00
Tennisclub.....	FL	335.00
Gemeente belast.....	FL	119.60
Onroer. goed bel.....	FL	131.00
Inboedelverzek.....	FL	195.25
M.R. particulier.....	FL	69.48
Gezinsongevallen.....	FL	93.50
Centraal Beheer.....	FL	437.50
Water.....	FL	444.00

TOTAAL FL 126627.52

W A S T E L A S T E N I N J A N .

Huur.....	FL	861.00
Gas.....	FL	268.00
Electriciteit.....	FL	234.00
Telefoon.....	FL	180.00
Dagblad.....	FL	63.61
Ziekenfonds.....	FL	15.00
Tennisclub.....	FL	335.00
Water.....	FL	111.00

TOTAAL FL 2067.61

```

10 REM *****
    ** VASTE LASTEN BEWAKING **
    ** (C)1985 RON TEN BERGE **
    ** VOOR MSX Aangepast **
    **DOOR:ROBERT-JAN DONKERS**
    *****
20 REM *****
    **      INITIALISEREN      **
    *****
30 CLS:WIDTH 40:COLOR 15,1,1:KEYOFF
40 N$(51),M$(51,12),G$(51),J$(51),I$(51)
  ,F(51)
50 FOR N=1 TO 51
60 N$(N)="*....."
70 NEXT N
80 REM *****
    **      HOOFDMENU      **
    *****
90 CLS
100 PRINT"                %%%%%%%%%%%
                                %           %
                                %  M E N U  %
                                %           %
                                %%%%%%%%%%%

110 PRINT:PRINT
120 PRINT:PRINT"          Maandoverzicht----
----->= 1"
130 PRINT:PRINT"          Jaaroverzicht----
----->= 2"
140 PRINT:PRINT"          Vaste lasten invoe
ren---->= 3"
150 PRINT:PRINT"          Gegevens afvoeren-
----->= 4"
160 PRINT:PRINT"          Gegevens opslaan--
----->= 5"
170 PRINT:PRINT"          Gegevens inladen--
----->= 6"
180 LOCATE 0,22:PRINT"          TOETS UW
  KEUZE IN !"
190 Q$=INKEY$:IF Q$="" THEN 190

```

```

200 IF Q$<"1" OR Q$>"6" THEN 190
210 ON VAL(Q$) GOTO 550,1110,220,830,127
0,1530
220 REM *****
** VASTE LASTEN INVOEREN **
*****

230 CLS
240 PRINT"          VASTE LASTEN INVOEREN "
:PRINT
250 PRINT"%%%%%%%%%"
:PRINT
260 FOR N=1 TO 50:IF N$(N)="*.....
...." THEN 290
270 NEXT N
280 PRINT:PRINT" HET BESTAND IS VOL !!!!
!!!":FOR Q=1 TO 1500:NEXT Q:GOTO 90
290 INPUT"OMSCHRIJVING   : ";Q$
300 IF Q$="" THEN LOCATE 0,5:GOTO 290
310 IF LEN(Q$)>16 THEN PRINT"OMSCHRIJVING
TE LANG (MAX.16) !":LOCATE 0,5:GOTO 29
0
320 LL=LEN(Q$)
330 MID$(N$(N),1,LL)=Q$
340 PRINT
350 INPUT"FREQUENTIE PER JAAR : ";F(N)
360 IF F(N)>12 THEN PRINT"REKEN BEDRAG O
M PER MAAND !!":GOTO 350
370 IF F(N)=12 THEN RESTORE:FOR I=1 TO 1
2:READ A$:M$(N,I)=A$:NEXT I:GOTO 440
380 PRINT:PRINT"IN DE MAAND(N):"
390 FOR T=1 TO F(N)
400 INPUT"MAANDNR.(1 T/M 12):";Q
410 RESTORE:FOR I=1 TO Q:READ K$:NEXT I:
M$(N,T)=K$
420 PRINT M$(N,T)
430 NEXT T
440 PRINT:INPUT"TERMIJNBEDRAG   : ";G$(N)
450 PRINT:INPUT"JAARAFREKENING(0=GEEN):"
;J$(N)
460 IF VAL( J$(N))=0 THEN 500

```



```

470 INPUT"IN WELKE MAAND ? (1 T/M 12) ";
I
480 IF I<1 OR I>12 THEN 470
490 RESTORE:FOR F=1 TO I:READ A$:NEXT F:
I$(N)=A$
500 LOCATE 0,22:PRINT"    GEGEVENS IN ORD
E ? (Ja/Nee) "
510 Q$=INKEY$:IF Q$="" THEN 510
520 IF Q$="J"OR Q$="j" THEN 80
530 IF Q$="N" OR Q$="n" THEN CLS:PRINT"
    VASTE LASTEN INVOEREN":PRINT:PRINT"%
%%%%%%%%%"
PRINT:GOTO 290
540 GOTO 510
550 REM *****
      **
      **    VASTE LASTEN PER    **
      **          MAAND          **
      *****
560 H=0:CLS
570 PRINT" VASTE LASTEN IN EEN BEPAALDE
MAAND ":PRINT
580 PRINT"%%%%%%%%%"
PRINT"%%%%%%%%%"
590 INPUT"WELKE MAAND (1 T/M 12) :";K
600 IF K<1 OR K>12 THEN 560
610 RESTORE:FOR N=1 TO K:READ K$:NEXT N
620 CLS:LOCATE 16,0:PRINT"MAAND:";K$:PRI
NT
630 FOR N=1 TO 50
640 FOR T=1 TO 12
650 IF N$(N)="*....." THEN 690
660 IF M$(N,T)=K$ THEN PRINT N$(N);;PRIN
T USING"F1 ####.##";VAL(G$(N)):H=H+VAL(
G$(N)):GOTO 680
670 NEXT T
680 NEXT N
690 KOP=0
700 FOR N=1 TO 50
710 IF N$(N)="*....."THEN 760

```

```

720 IF I$(N)<>K$ THEN 750
730 IF KOP=0 THEN PRINT:PRINT"JAARAFREKE
NING:":LET KOP=1
740 PRINT N$(N);:PRINT USING"F1 #####.##
";VAL(J$(N)):H=H+VAL(J$(N))
750 NEXT N
760 PRINT:PRINT"%%%%%%%%%%%%%%
%%%%%%%%%%%%%":PRINT
770 PRINT "TOTAAL          ";:PRINT USIN
G"F1#####.##";H
780 LOCATE 0,22:PRINT"          >M< =MENU    >A
< = ANDERE MAAND"
790 Q$=INKEY$:IF Q$="" THEN 790
800 IF Q$="M" OR Q$="m" THEN 80
810 IF Q$="A" OR Q$="a" THEN 550
820 GOTO 790
830 REM *****
** GEGEVENS UIT HET BESTAND **
**          AFVOEREN          **
*****

840 CLS
850 LOCATE 0,21:PRINT"          >> M <<
= MENU"
860 LOCATE 0,3:INPUT"WAT WILT U AFVOEREN
?";K$
870 IF K$="M" OR K$="m" THEN 80
880 LL=LEN(K$):IF LL>16 THEN 840
890 LOCATE 0,21:PRINT SPACE$(38)
900 FOR N=1 TO 50:IF K$=MID$(N$(N),1,LL)
THEN 930
910 NEXT N
920 LOCATE 0,3:PRINT K$;"ZIT NIET IN HET
BESTAND !":FOR V=1 TO 1500:NEXT V:GOTO
830
930 LOCATE 0,3:PRINT"OMSCHRIJVING :";N$(
N)
940 LOCATE 0,21:PRINT" TOETS IN .....>W
< =WISSEN >M< = MENU"
950 Q$=INKEY$:IF Q$="" THEN 950
960 IF Q$="M" OR Q$="m" THEN 80

```

```

970 IF Q$="W" OR Q$="w" THEN 990
980 GOTO 950
990 REM *****
    **
    **      GEGEVENS WISSEN      **
    **
    *****
1000 FOR T=1 TO 51:IF N$(T)="*....."
    ...." THEN 1020
1010 NEXT T
1020 N$(N)=N$(T-1):N$(T-1)=N$(T):F(N)=F(
T-1):F(T-1)=F(T)
1030 FOR M=1 TO 12:M$(N,M)=M$(T-1,M):M$(
T-1,M)=M$(T,M):NEXT M
1040 G$(N)=G$(T-1):G$(T-1)=G$(T):J$(N)=J
$(T-1):J$(T-1)=J$(T)
1050 I$(N)=I$(T-1):I$(T-1)=I$(T)
1060 LOCATE 4,10:PRINT" DE GEGEVENS ZIJN
    GEWIST !"
1070 Q$=INKEY$:IF Q$="" THEN 1070
1080 IF Q$="M" OR Q$="m" THEN 80
1090 IF Q$="W" OR Q$="w" THEN 830
1100 GOTO 1070
1110 REM *****
    **
    ** J A A R O V E R Z I C H T **
    **
    *****
1120 CLS
1130 PRINT" J A A R O V E R Z I C H T"
1140 PRINT
1150 TT=0
1160 FOR N=1 TO 51
1170 H=0
1180 IF N$(N)="*....."THEN 123
0
1190 H=H+(F(N)*VAL(G$(N)))+VAL(J$(N))
1200 PRINT N$(N);:PRINT USING"F1 #####.#
#":H
1210 TT=TT+H

```

```

1220 NEXT N
1230 PRINT:PRINT"JAARTOTAAL          ";:PRIN
T USING"F1#####.##";TT
1240 LOCATE 0,22:PRINT"          >>> M
    <<< = MENU"
1250 Q$=INKEY$:IF Q$="" THEN 1250
1260 IF Q$="M" OR Q$="m" THEN 80 ELSE 12
50
1270 REM *****
    **
    **      GEGEVENS      OPSLAAN      **
    **
    *****

1280 CLS
1290 PRINT:PRINT"MET WELKE BAUDRATE WILT
U NAAR CASSETTE SCHRIJVEN ? 1=1200 2=24
00 BAUD "
1300 Q$=INKEY$:IF Q$="" THEN 1300
1310 IF Q$="1" THEN BR=1:GOTO 1340
1320 IF Q$="2" THEN BR=2:GOTO 1340
1330 GOTO 1300
1340 PRINT:PRINT"GEEF NAAM VAN DE FILE:(
MAX.6 LETTERS)
1350 SCREEN,,,BR
1360 INPUT Q$
1370 S$="CAS:"+Q$
1380 PRINT"ZET CAS.RECORDER KLAAR (RECOR
D & PLAY)"
1390 PRINT"TOETS DAARNA RETURN VOOR SAVE
N"
1400 Q$=INKEY$:IF ASC(Q$)<>13 THEN 1400
1410 OPEN S$ FOR OUTPUT AS#1
1420 MOTOR ON
1430 PRINT#1,N
1440 FOR K=1 TO N:PRINT#1,N$(K)
1450 PRINT#1,F(K)
1460 PRINT#1,G$(K)
1470 PRINT#1,I$(K)
1480 PRINT#1,N$(K):NEXT K
1490 FOR K=1 TO N:FOR M=1 TO 12:PRINT#1,
M$(K,M):NEXT M:NEXT K

```

```

1500 CLOSE #1
1510 MOTOR OFF
1520 GOTO 80
1530 REM *****
      **
      **   GEGEVENS   INLADEN   **
      **
      *****
1540 CLS
1550 PRINT:PRINT"GEEF NAAM VAN DE FILE:(
MAX.6 LETTERS)
1560 INPUT Q$
1570 S$="CAS:"+Q$
1580 PRINT"ZET CAS.RECORDER KLAAR (PLAY)
"
1590 PRINT"TOETS DAARNA RETURN VOOR LADE
N"
1600 Q$=INKEY$:IF ASC(Q$)<>13 THEN 1400
1610 OPEN S$ FOR INPUT AS#1
1620 MOTOR ON
1630 INPUT#1,N
1640 FOR K=1 TO N:INPUT#1,N$(K)
1650 INPUT#1,F(K)
1660 INPUT#1,G$(K)
1670 INPUT#1,I$(K)
1680 INPUT#1,N$(K):NEXT K
1690 FOR K=1 TO N:FOR M=1 TO 12:INPUT#1,
M$(K,M):NEXT M:NEXT K
1700 CLOSE #1
1710 MOTOR OFF
1720 GOTO 80
1730 DATA "JAN.", "FEB.", "MRT.", "APR.", "M
EI.", "JUN.", "JUL.", "AUG.", "SEP.", "OKT.",
"NOV.", "DEC."

```

FAKTUREREN

Faktureren is een simpel doch efficiënt programma, dat voor U fakturen maakt inclusief alle daarvoor benodigde berekeningen. Het ontwerp van de faktuur kunt U aan Uw eigen wensen aanpassen. Houdt bij het aanpassen rekening met de volgende punten.

1. Het adres van de klant staat op de juiste plaats voor het gebruik van standaard vensterenveloppen, verander daar dus niets aan.
2. Dit programma gaat ervan uit, dat alle artikelen in DATA-regels zijn opgenomen. De artikelnummer corresponderen met deze dataregels en wel als volgt: De artikelen zijn genummerd van 1001 en verder. De RESTORE opdracht in regel 680 verwijst dan naar de dataregels 2250 en verder. Afhankelijk van het aantal artikelen moet U ook nog de regels 660 en 670 aanpassen.
3. In de DATA regels vult U achtereenvolgens de volgende gegevens van het artikel in:
 - Artikelnummer, - Artikelnaam, - Stuksprijs excl. BTW en de - BTW code (1 = 5%, 2 = 19%, 0 = 0%). Alles tussen de bekende aanhalingstekens. De computer leest alles in Strings in.
4. In het programma is een variabele R opgenomen, deze R is de printregelteller. Op de meeste Printers kunt U op een A4 72 regels printen. Daarom is de regelteller ingevoerd; heeft de regelteller een bepaalde waarde bereikt, dan wordt er een subtotaal gemaakt en getransporteerd naar de vervolgfaktuur.

Het programma is van REM-statements voorzien om een duidelijk onderscheid te maken tussen de diverse routines.

TERMINAL SOFTWARE PUBLICATIES

Postbus 111
5110 AC Baarle-Nassau
Tel. 04257-9161

Bankrelatie:.....
Kamer van Koophandel:.....

Faktuurnummer : 84510

De Gouden Doortrapper B.V.
Pedaallaan 2
9999 XX Fietsdorp

Baarle-Nassau.17 NOVEMBER 1984

AANIAL	ART.NR.	OMSCHRIJVING	PRIJS EXCL.	BTW	NETTO
3	1005	Bankschroef 28mm	F1. 38.50	1	F1. 115.50
4	1004	Steekslutrel 10/11	F1. 7.60	1	F1. 30.40
2	1007	Schroevendraaier 4mm	F1. 6.19	1	F1. 12.38
4	1008	Sikkels 50cm 8 st.	F1. 45.17	1	F1. 180.68
3	1001	BOUTEN M8x30 50 st.	F1. 12.45	1	F1. 37.35
20	1003	VERPAKKINGSDOZEN 50	F1. 24.95	2	F1. 499.00
					+
SUBTOTAAL EXCL.:					F1. 875.31
BTW TOTAAL					:F1. 113.63
					+
TOTAAL					:F1. 988.94
Kredietbeperking 2 %					19.78
					+
FAKTUURBEDRAG					:F1. 1008.72

BTW 1 (5%) :F1.18.82 BTW 2 (19%) :F1.94.81

Betaling 8 dagen netto

```

10 REM *****
    **
    ** FAKTUREREN MET DE MSX **
    **
    *****
20 REM ** ALLEEN VOOR 80-KOLOMS **
    **          PRINTERS          **
    *****
30 CLS:WIDTH 40:COLOR 15.1,1:KEYOFF
40 REM *****
    **          INITIALISEREN          **
    *****
50 K1=0:K2=0:T=0
60 N=100:REM AANTAL ARTIKELEN
70 DIM A(N):REM AANTAL PER ARTIKEL
80 DIM A$(N):REM ARTIKELNUMMERS
90 DIM B$(N):REM ARTIKELNAAM
100 DIM C$(N):REM PRIJS
110 DIM D$(N):REM BTW-CODE
120 DIM E$(N):REM AANTAL x PRIJS
130 REM*****
    ** UW FIRMANAAM + ADRES **
    *****
140 N$="TERMINAL SOFTWARE PUBLICATIES"
150 O$="Postbus 111"
160 P$="5110 AC BAARLE NASSAU"
170 R$="Tel:04257-9161"
180 S$="Bankrelatie....."
190 T$="Kamer van Koophandel....."
200 U$="Betaling 8 dagen NETTO"
210 V$="Kredietbeperking 2% "
220 INPUT"FAKTUURDATUM : ";F$
230 PRINT
240 INPUT"LAATSTE FAKTUURNO. :";F1
250 PRINT:PRINT"ALLES IN ORDE ? (Ja/Nee)
"
260 Q$=INKEY$:IF Q$="" THEN 260
270 IF Q$="J" OR Q$="j" THEN 300
280 IF Q$="N" OR Q$="n" THEN CLS:GOTO 22
0

```



```

290 GOTO 260
300 REM *****
      **                                     **
      ** OPSTELLEN VAN FAKTUUR **
      **                                     **
      *****
310 CLS:COLOR 15,12,12
320 PRINT
330 INPUT"NAAM KLANT      ":";G$
340 INPUT"ADRES          ":";H$
350 INPUT"P.CODE+PLAATS ":";I$
360 PRINT:PRINT
370 PRINT"FAKTUURNO.     ":";F1+1
380 PRINT"FAKTUURDATUM  ":";F$
390 LOCATE 0,22:PRINT"      GEGEVENS IN
ORDE ? (Ja/Nee)      "
400 Q$=INKEY$:IF Q$="" THEN 400
410 IF Q$="J" OR Q$="j" THEN 440
420 IF Q$="N" OR Q$="n" THEN 310
430 GOTO 400
440 REM *****
      **                                     **
      **      HOOFDPROGRAMMA      **
      **                                     **
      *****
450 F1=F1+1:GOSUB 480
460 K1=0:K2=0:T=0:GOSUB 730
470 GOTO 940
480 REM *****
      ** ARTIKEL ZOEKEN IN DATA**
      *****
490 C=1
500 CLS
510 INPUT"GEEF ARTIKELNUMMER ":";QQ
520 IF QQ<1001 THEN PRINT"Nummer te klein !!":FOR Q=1 TO 1000:NEXT Q:GOTO 500
530 IF QQ>1008 THEN PRINT"Nummer te groot!!":FOR Q=1 TO 1000:NEXT Q:GOTO 500
540 RESTORE
550 FOR Q=1 TO (QQ-1000)

```

```

560 READ A$(C),B$(C),C$(C),D$(C)
570 NEXT Q
580 PRINT:PRINT A$(C);" ";B$(C)
590 PRINT:INPUT"AANTAL STUKS      ":A(C)
600 IF A(C)<1 THEN 590
610 EE=A(C)*VAL(C$(C))
620 E$(C)=STR$(EE)
630 LOCATE 0,22:PRINT"          GEGEVENS IN O
RDE (Ja/Nee)      "
640 Q$=INKEY$:IF Q$="" THEN 640
650 IF Q$="J" OR Q$="j" THEN 680
660 IF Q$="N" OR Q$="n" THEN 500
670 GOTO 640
680 LOCATE 0,22:PRINT" NOG EEN ARTIKEL V
OOR DEZE KLANST (J/N)"
690 Q$=INKEY$:IF Q$="" THEN 690
700 IF Q$="N" OR Q$="n" THEN RETURN
710 IF Q$="J" OR Q$="j" THEN 720 ELSE 69
0
720 C=C+1:GOTO 500
730 REM *****
      **
      ** KOP VAN FAKTUUR PRINTEN **
      **
      *****
740 LPRINT
750 LPRINT CHR$(27);"H";CHR$(14);TAB(6);
N$;CHR$(15)
x 760 LPRINT:LPRINT CHR$(27);"N"
770 LPRINT O$;TAB(45-LEN(O$));S$
780 LPRINT P$;TAB(45-LEN(P$));T$
790 LPRINT R$
800 LPRINT:LPRINT
810 LPRINT"FAKTUURNUMMER: ";F1
820 LPRINT:LPRINT:LPRINT
830 LPRINT TAB(43);G$
840 LPRINT TAB(43);H$
850 LPRINT TAB(43);I$
860 LPRINT"BAARLE NASSAU,";F$
870 LPRINT

```

```

880 LPRINT
890 LPRINT "AANTAL      ART.NO.      OMSCHRIJVING
      PRIJS EXCL.      BTW      N
      ETTO"
900 LPRINT "=====
=====
=====
910 LPRINT
920 R=25
930 RETURN
940 REM *****
      **
      **      PRINTROUTINE      **
      **
      *****
950 FOR Q=1 TO C
960 X$=STR$(A(Q)):L5=LEN(X$):L1=LEN(A$(Q)):L2=LEN(B$(Q)):L3=L1+L2+L5:L4=L3+L5+10
970 LPRINT A(Q);TAB(5-L5);A$(Q);TAB(5-L5);B$(Q);TAB(38-L3);:LPRINT USING "F1 ####.##":VAL(C$(Q));:LPRINT TAB(7);D$(Q);:LPRINT USING "      F1 #####.##";VAL(E$(Q))
980 R=R+1:IF R=60 THEN GOSUB 1110
990 IF Q=C THEN 1010
1000 NEXT Q
1010 GOSUB 1110
1020 REM *****
      **
      **      EINDE VAN FAKTUUR      **
      **
      *****
1030 CLS
1040 LOCATE 3,9:PRINT "'1' = Nog een exemplaar"
1050 LOCATE 3,11:PRINT "'2' = Een nieuwe faktuur"
1060 LOCATE 0,22:PRINT "      TOETS UW KEUZE IN (1/2)      "
1070 Q$=INKEY$:IF Q$="" THEN 1070
1080 IF Q$="1" THEN 460

```

```

1090 IF Q$="2" THEN 440
1100 GOTO 1070
1110 REM*****
      **
      **      SUBTOTAALTELLER      **
      **
      *****

```

```

X 1120 FOR L=1 TO Q
1130 T=T+VAL(E$(L))
1140 IF D$(L)="1" THEN W=5:GOTO 1170
1150 IF D$(L)="2" THEN W=19:GOTO 1180
7365 X 1160 W=0:NEXT L:GOTO 1170.
1170 K1=K1+(W/100)*VAL(E$(L)):GOTO 1190
1180 K2=K2+(W/100)*VAL(E$(L))
X 1190 NEXT L
1200 IF R<60 THEN 1270
1210 LPRINT TAB(48);"=====
=====
1220 LPRINT:LPRINT TAB(46);"TE TRANSPORT
EREN:";USING" F1 #####.##";T
1230 FOR M=1 TO (72-R)
1240 LPRINT :NEXT M
1250 GOSUB 730
1260 RETURN
1270 LPRINT TAB(67);"+ -----"
1280 LPRINT TAB(52);"SUBTOTAAL EXCL.":;L
PRINT USING" F1 #####.##";T
1290 BT=K1+K2:BT=INT(BT*100+.5)/100
1300 LPRINT TAB(52);"BTW TOTAAL "":;L
PRINT USING" F1 #####.##";BT
1310 LPRINT TAB(67);"+ -----"
1320 LPRINT TAB(52);"TOTAAL "":;L
PRINT USING" F1 #####.##";(BT+T)
1330 KP=INT(.02*(BT+T)*100+.5)/100
1340 LPRINT TAB(50);"KREDIETBEPERKING ";
USING" F1 #####.##";KP
1350 LPRINT TAB(67);"+ -----"
1360 LPRINT TAB(52);"FAKTUURBEDRAG "":;L
PRINT USING" F1#####.##";(BT+T+KP)
1370 LPRINT:LPRINT

```

```

1380 K1=INT(K1*100+.5)/100
1390 K2=INT(K2*100+.5)/100
1400 LPRINT" BTW 1 (5%):";USING"F1 ####
.##":K1::LPRINT" BTW 2 (19%):"; USING"F
1 ####.##":K2
1410 LPRINT:LPRINT U$
1420 FOR P=1 TO (72-R):LPRINT:NEXT P
1430 RETURN
1440 REM *****
**
** DATA - REGELS **
**
** *****
1450 REM ** **
** VOER HIER UW EIGEN **
** ARTIKELEN IN **
** OP VOLGORDE VAN **
1460 REM ** ARTIKELNUMMERS **
** *****
1470 REM ** **
** HOUDT DAARBIJ DE **
** VOLGENDE VOLGORDE AAN **
**
1480 REM ** 1.ARTIKELNUMMER **
** 2.OMSCHRIJVING **
** 3.VERKOOPSPRIJS EXCL. **
** 4.BTW CODE **
1490 REM ** 1= 5% 2= 19% **
**
** *****
1500 DATA "1001","BOUTEN M8x30 50 STUKS"
,"12.45","2"
1510 DATA "1002","MOEREN M8 50 STUKS"
,"6.10","2"
1520 DATA "1003","STEEKSLEUTEL 10/11 3 S
TUKS","9.65","2"
1530 DATA "1004","SCHROEVENDRAAIER 6 mm"
,"6.19","2"
1540 DATA "1005","GEREEDSCHAPSKIST METAAL"
,"16.95","2"

```

1550 DATA "1006","BANKSCHROEF 50 mm","38
.50","2"
1560 DATA "1007","SIKKELS 50 cm 8 STUKS"
."45.17","2"
1570 DATA "1008","HANDLEIDING AEG GEREED
SCHAP","16.75","1"
1580 DATA "ZZZZ","Z","Z","Z"

PRIJSLIJSTEN

Dit programma maakt voor U een verkoopsprijslijst uitgaande van Uw inkoopprijzen.

Het programma genereert een prijslijst voor 50 artikelen. Wilt U meer artikelen in Uw prijslijst opnemen dan past U de DIM-statements in regels 60-100 aan.

Runt U het programma dan wordt U eerst gevraagd welke Brutomarge U wilt behalen. Deze brutomarge wordt berekend van de uiteindelijke verkoopsprijs. Stel dat U een Brutomarge van 30% wilt, dan berekent het programma in de regels 810-840 de verkoopsprijs uitgaande van de inkoopsprijs.

Nemen we bijv. Uw inkoopsprijs van f 10,00 excl. BTW dan wordt met een brutomarge van 30% Uw verkoopsprijs f 14,29 excl. BTW. Uw brutomarge is dan $30/100 \times 14,29 = f$ 4,29.

Nadat U de marge heeft ingevoerd begint het programma.

U vult alle artikelen in de prijslijst in. Tijdens het invoeren kunt U steeds met printen beginnen. Deze printroutine start op regel 900.

Een nadeel van veel computers is dat wanneer ze getallen moeten afdrukken met een precisie van twee cijfers achter de komma, de laatste nul wegvalt. Met MSX-Basic heeft U daar gelukkig geen last van. In het programma is dan ook gebruik gemaakt van de opdracht PRINT USING.

```

10 REM *****
   **
   ** PROGRAMMA VOOR HET **
   **     MAKEN VAN     **
   **     PRIJSLIJSTEN  **
20 REM **
   *****
30 CLS:WIDTH 40:COLOR 15,4,4:KEYOFF
40 REM *****
   **
   **VOOR 80 KOLOMS PRINTER**
   **
   *****
50 REM **
   **     INITIALISEREN  **
   **
   **
   *****
60 DIM A$(50):REM ARTIKELNAAM
70 DIM P$(50):REM VERKOOPSPRIJS/STUK
   EXCL. BTW
80 DIM S$(50):REM SPECIALE VERKOOPS-
   PRIJS/STUK
90 DIM B(50): REM BTW TARIEF
100 DIM I$(50):REM INKOOPSPRIJS/STUK
   EXCL. BTW
110 S=0
120 INPUT"HOEVEEL % BRUTOMARGE WILT U :";B1
130 PRINT
140 PRINT"HANTEERT U SPECIALE VERKOOPSPRIJZEN BIJ AFNAME VAN AANTALLEN (Ja/Nee)";
:PRINT
150 Q$=INKEY$:IF Q$="" THEN 150
160 IF Q$="J" OR Q$="j" THEN S=1:GOTO 190
170 IF Q$="N" OR Q$="n" THEN 220
180 GOTO 150
190 INPUT"BIJ WELK AANTAL GEEFT U EEN SPECIALE VERKOOPSPRIJS :";Z1
200 PRINT:INPUT"WAT IS DAN DE EXTRA KORTING IN % ";Y1

```



```

210 GOTO 1310
220 REM *****
      **                                     **
      **   INVOEREN ARTIKELEN   **
      **   MET HUN INKOOPSPRIJS **
230 REM **                                     **
      *****
240 CLS:N=N+1
250 PRINT"Artikelnummer      :";N
260 PRINT
270 INPUT"Artikelnaam      :";K$
280 IF LEN(K$)>25 THEN CLS:GOTO 250
290 A$(N)=K$
300 PRINT:INPUT"Inkoopprijs      :";I$(N)
)
310 PRINT:PRINT"IS DE INKOOPSPRIJS (Excl
.of Incl.) "
320 Q$=INKEY$:IF Q$="" THEN 320
330 IF Q$="E" OR Q$="e" THEN B$="Excl.":
C=0:GOTO 410
340 IF Q$="I" OR Q$="i" THEN B$="Incl.":
C=1:GOTO 360
350 GOTO 320
360 PRINT:PRINT" WAT IS HET BTW-TARIEF (
1= 5% ,2=19%) "
370 Q$=INKEY$:IF Q$="" THEN 370
380 IF Q$="1" THEN B(N)=5:GOTO 420
390 IF Q$="2" THEN B(N)=19:GOTO 420
400 GOTO 370
410 LOCATE 0,10:
420 IF C=1 THEN GOSUB 520
430 GOSUB 560
440 PRINT:PRINT"VERKOOPSPRIJS :";USING"F
1 #####.##";VAL(P$(N))
450 IF S=1 THEN GOSUB 600
460 IF S=1 THEN PRINT:PRINT"SPECIALE VER
KOOPSPRIJS:";USING"F1 #####.##";VAL(S$(N)
))
470 LOCATE 0,22:PRINT" A=NOG EEN ARTIKEL
M=MENU "

```

```

480 Q$=INKEY$:IF Q$="" THEN 480
490 IF Q$="A" OR Q$="a" THEN 220
500 IF Q$="M" OR Q$="m" THEN 1310
510 GOTO 480
520 I=(VAL(I$(N))*100)/(100+B(N))
530 I$(N)=STR$(INT(I*100+.5)/100)
540 PRINT"INKOOPSPRIJS excl.:";USING"F1
####.##";VAL(I$(N))
550 RETURN
560 M=INT(((100-B1)/100)*100+.5)/100
570 P=INT((VAL(I$(N))/M)*100+.5)/100
580 P$(N)=STR$(P)
590 RETURN
600 REM *****
      **
      ** SPECIALE VERKOOPSPRIJS **
      **
      *****
610 P=VAL(P$(N))-(INT(((Y1/100)*VAL(P$(N)
)))*100+.5)/100)
620 S$(N)=STR$(P)
630 RETURN
640 REM *****
      **
      **          PRINT-ROUTINE          **
      **
      *****
650 CLS:COLOR 15,1,1
660 IF E=0 THEN PRINT"EERST PRIJSLIJST O
PSTELLEN":FOR J=1 TO 1500:NEXT J:GOTO 13
10
670 PRINT"  PRIJSLIJSTEN PRINTEN"
680 PRINT:INPUT"HOEVEEL PRIJSLIJSTEN WIL
T U PRINTEN ";X
690 PRINT:PRINT"KOP ZIET ER ALS VOLGT UI
T:":PRINT"=====
=====":PRINT
700 PRINT E$:REM UW FIRMANAAM
710 PRINT F$:REM ADRES
720 PRINT G$:REM P.CODE+PLAATS

```

```

730 PRINT "PRIJSLIJST PER:";D$:REM DATUM
740 PRINT:PRINT
750 PRINT "ER ZIJN ";N;" ARTIKELEN INGEVO
ERD "
760 LOCATE 0,22:PRINT "A= ARTIKELLIJST M
=MENU P=PRINTEN "
770 Q$=INKEY$:IF Q$="" THEN 770
780 IF Q$="M" OR Q$="m" THEN 1310
790 IF Q$="A" OR Q$="a" THEN 820
800 IF Q$="P" OR Q$="p" THEN 910
810 GOTO 770
820 REM *****
      **
      ** ARTIKELLIJST OP SCHERM **
      **
      *****
830 CLS:COLOR 1,4,4
840 FOR M=1 TO N
850 PRINT M;"= ":A$(M)
860 PRINT "VERKOOPSPRIJS:";USING"F1 ####
.##";VAL(P$(M))
870 PRINT "SPECIALE PRIJS:";USING"F1 ####
#.##";VAL(S$(M));:PRINT " per:";Z1
880 PRINT
890 NEXT M
900 GOTO 760
910 REM *****
      **
      ** PRIJSLIJST PRINTEN **
      **
      *****
920 CLS:COLOR 15,1,1
930 R=0
940 FOR L=1 TO X
950 LPRINT CHR$(27);"H":REM ZIE HANDLEID
ING VAN UW EIGEN PRINTER (LETTERTYPE)
960 LPRINT CHR$(14);E$;CHR$(15)
970 LPRINT F$
980 LPRINT G$
990 LPRINT

```

```

1000 LPRINT"PRIJSLIJST PER:";D$
1010 LPRINT
1020 LPRINT"=====
=====
=====
1030 LPRINT"ARTIKEL
VERKOOP PER:";Z1
1040 LPRINT"
EXCL. EXCL."
1050 LPRINT"=====
=====
=====
1060 LPRINT
1070 R=12
1080 FOR M=1 TO N
1090 LPRINT A$(M);TAB(60-LEN(A$(N)));:LP
RINT USING"F1 #####.##";VAL(P$(M));:LPRI
NT USING" F1 #####.##";VAL(S$(M))
1100 R=R+1
1110 NEXT M
1120 FOR P=1 TO (70-R):LPRINT:NEXT P
1130 NEXT L
1140 GOTO 1310
1150 REM *****
**
** KOP VAN FAKTUUR INVOEREN **
**
*****
1160 CLS:COLOR 4,7,7
1170 PRINT" VERSTREK DE VOLGENDE GEGEVEN
S !":PRINT
1180 PRINT"%%%%%%%%%%%%%%
%%%%%%%%%%":PRINT
1190 INPUT"UW FIRMANAAM :";E$
1200 INPUT"ADRES :";F$
1210 INPUT"P.CODE+PLAATS :";G$
1220 PRINT
1230 INPUT"PER WELKE DATUM IS DE PRIJSLI
JST GELDIG: ";D$
1240 PRINT

```

```

1250 INPUT"GEEF DATUM OOK IN VOLGENDE NO
TATIE: (DDMMJJ) ! ";X$
1260 LOCATE 0,22:PRINT" ALLE GEGEVEN
S IN ORDE (Ja/Nee) ?"
1270 Q$=INKEY$:IF Q$="" THEN 1270
1280 IF Q$="J" OR Q$="j" THEN E=1:GOTO 1
310
1290 IF Q$="N" OR Q$="n" THEN 1150
1300 GOTO 1270
1310 REM *****
** **
** M E N U **
** **
*****
1320 CLS:COLOR 15,12,12
1330 LOCATE 15,0:PRINT" M E N U "
1340 PRINT:PRINT"=====
=====":PRINT
1350 PRINT" 1 >=====PRIJSLIJST OP
STELLEN":PRINT
1360 PRINT" 2 >=====INKOOPSPRIJZEN I
NVOEREN":PRINT
1370 PRINT" 3 >=====PRINTEN VAN PRIJS
LIJSTEN":PRINT
1380 PRINT" 4 >=====PRIJSLIJST OP TAPE
OPSLAAN":PRINT
1390 PRINT" 5 >=====PRIJSLIJST LADEN V
AN TAPE":PRINT
1400 LOCATE 0,22:PRINT" TOETS UW
KEUZE IN"
1410 Q$=INKEY$:IF Q$="" THEN 1410
1420 IF Q$<"1" OR Q$>"5" THEN 1410
1430 Q=VAL(Q$)
1440 ON Q GOTO 1150,220,640,1460,1680
1450 GOTO 1410
1460 REM *****
** **
** GEGEVENS OP TAPE OPSLAAN **
** **
*****

```

```

1470 CLS:COLOR 1,11,11
1480 PRINT"GEGEVENS WORDEN WEGGESCHREVEN
      NAAR TAPE ONDER DE FILENAAM:";X$
1490 LOCATE0,22:PRINT" TOETS > RETURN <
      ALS U WILT BEGINNEN "
1500 Q$=INKEY$:IF Q$="" THEN 1500
1510 IF ASC(Q$)<>13 THEN 1500
1520 MOTOR ON
1530 OPEN"CAS:"X$ FOR OUTPUT AS #1
1540 PRINT#1,N
1550 FOR M=1 TO N
1560 PRINT#1,A$(M)
1570 PRINT#1,P$(M)
1580 PRINT#1,S$(M)
1590 PRINT#1,I$(M)
1600 NEXT M
1610 PRINT#1,E$
1620 PRINT#1,F$
1630 PRINT#1,G$
1640 PRINT#1,D$
1650 CLOSE#1
1660 MOTOR OFF
1670 GOTO 1310
1680 REM *****
      ** **
      ** GEGEVENS VAN TAPE LADEN **
      ** **
      *****
1690 CLS:COLOR 1,11,11
1700 INPUT"WELKE FILE MOET VAN TAPE GELA
      DEN WORDEN";X$
1710 LOCATE0,22:PRINT" TOETS > RETURN <
      ALS U WILT BEGINNEN "
1720 Q$=INKEY$:IF Q$="" THEN 1500
1730 IF ASC(Q$)<>13 THEN 1500
1740 MOTOR ON
1750 OPEN"CAS:"X$ FOR INPUT AS #1
1760 INPUT#1,N
1770 FOR M=1 TO N
1780 INPUT#1,A$(M)

```

```
1790 INPUT#1,P$(M)
1800 INPUT#1,S$(M)
1810 INPUT#1,I$(M)
1820 NEXT M
1830 INPUT#1,E$
1840 INPUT#1,F$
1850 INPUT#1,G$
1860 INPUT#1,D$
1870 CLOSE#1
1880 MOTOR OFF
1890 E=1:GOTO 1310
```

FINANCIËLE BEDRIJFSANALYSE

Met dit programma kunt U als zakenman zich snel een beeld verschaffen van de financiële verhoudingen binnen Uw bedrijf.

Het programma verschaft U een aantal financiële kengetallen waarmee U Uw bedrijf kunt vergelijken met bedrijven in dezelfde branche. Deze laatste gegevens kunt U via de kamer van koophandel of bepaalde branche-gerichte verenigingen verkrijgen.

Ziet de vergelijking er voor Uw bedrijf niet zo zonnig uit dan wordt het tijd dat U Uw beleid eens aanpast. Met dit programma is het voor U een kleinigheid de zere plekken in Uw bedrijf te ontdekken.

Voor een optimaal gebruik van dit programma dient U over de volgende gegevens te beschikken:

- Saldo liquide middelen (Kas-Bank en Giro)
- Waarde van de voorraad
- Waarde van inventaris, gebouwen enz.
- Totaal van de vorderingen
- Totaal van de schulden op korte termijn
- Totaal van de schulden op lange termijn
- Totaal aan eventuele leningen
- Het eigen kapitaal van de onderneming
- Het reservekapitaal van de onderneming
- Onverdeelde winsten
- Bedrag van de kasverkopen
- Bedrag van verkopen op rekening
- Bedrag van afschrijvingen
- Bedrag van de inkopen
- Bedrag van de rente op eventuele lening
- Bedrag van huur of huurwaarde als U een eigen pand bezit
- Bedrag aan belastingen
- Totaal van de overige bedrijfskosten (kantoorkosten, verkoopkosten, enz)

Al deze gegevens kunt U vinden op de balans en de Winst- en Verliesrekening. De berekeningen van de diverse kengetallen zijn gebaseerd op het DU-PONT-SCHEMA.

BALANS OPSTELLING.

EIGEN VERMOGEN: -

KAPITAAL	1000
RESERVES	1000
ONVERD. WINST	1000
OVERIG	1000

4000

BESTAAT UIT: -

=====

VOORRAAD	2000
LIQ. MIDDELEN	1000
VORDERINGEN	2000
OV. VLOT. MIDDELEN	1000
FASTG. MIDDELEN	1000

7000

VERMINDERD NET: -

BANKLENINGEN	1000
SCHULDEN KORT	1000
SCHULDEN LANG	1000

3000

4000

=====

VERLIES & WINSTREKENING

ONTVANGSTEN

VERKOOP KONTANT	1000
VERKOOP OP REK.	1000
OV. INKOMSTEN	1000

MINUS KOSTEN

INKOPEN/AFSCHR.	1000
RENTE	1000
HUUR	1000
BELASTING	1000
OV. BEDRIJFSKOSTEN	1000

NETTO WINST/VERLIES

-2000

=====

```

10 REM *****
    **
    **      BEREKENING VAN DE      **
    ** FINANCIELE KENGETALLEN **
20 REM **      VOLGENS HET      **
    **      DU-PONT SCHEMA      **
    **
    *****
30 CLS:WIDTH 40:COLOR 15,4.4:KEYOFF
40 REM *****
    **      INITIALISEREN      **
    *****
50 DIM A(5),B(3),C(4),D(3),E(5)
60 DIM A$(5),B$(3),C$(4),D$(3),E$(5)
70 A=0:B=0:C=0:D=0:E=0:I=0:J=1
80 RESTORE
90 FOR F=1 TO 5:READ A$(F):NEXT F
100 FOR F=1 TO 3:READ B$(F):NEXT F
110 FOR F=1 TO 4:READ C$(F):NEXT F
120 FOR F=1 TO 3:READ D$(F):NEXT F
130 FOR F=1 TO 5:READ E$(F):NEXT F
140 GOSUB 630
150 GOTO 1520
160 REM *****
    **
    **      INVOEREN VAN GEGEVENS  **
    **
    *****
170 CLS:COLOR 1,11,11:PRINT
180 PRINT"      VOER DE VASTE GEGEVENS I
N !"
190 PRINT"^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^
^^^^^^^^^^^^^^^^"
200 PRINT"1.BEZITTINGEN:"
210 FOR X=1 TO 5
220 PRINT"----";A$(X);" ";
230 INPUT X$
240 GOSUB 760:IF I=0 THEN 220
250 A(X)=VAL(X$):A=A+A(X)

```

```

260 NEXT X
270 PRINT:PRINT"2.SCHULDEN : "
280 FOR X=1 TO 3
290 PRINT"----";B$(X);" ";
300 INPUT X$
310 GOSUB 760:IF I=0 THEN 290
320 B(X)=VAL(X$):B=B+B(X)
330 NEXT X
340 PRINT:PRINT"3.EIGEN VERMOGEN : "
350 FOR X=1 TO 4
360 PRINT"----";C$(X);" ";
370 INPUT X$
380 GOSUB 760:IF I=0 THEN 360
390 C(X)=VAL(X$):C=C+C(X)
400 NEXT X
410 PRINT:PRINT"4.OPBRENGSTEN : "
420 FOR X=1 TO 3
430 PRINT"----";D$(X);" ";
440 INPUT X$
450 GOSUB 760:IF I=0 THEN 430
460 D(X)=VAL(X$):D=D+D(X)
470 NEXT X
480 PRINT:PRINT"5.UITGAVEN      : "
490 FOR X=1 TO 5
500 PRINT"----";E$(X);" ";
510 INPUT X$
520 GOSUB 760:IF I=0 THEN 500
530 E(X)=VAL(X$):E=E+E(X)
540 NEXT X
550 I=1
560 IF A=B+C THEN RETURN
570 PRINT:PRINT" BALANS IS NIET IN EVENW
    ICHT !!!!!"
580 PRINT
590 PRINT" DRUK OP EEN TOETS VOOR KORREK
    TIE "
600 Q$=INKEY$:IF Q$="" THEN 600
610 GOSUB 850
620 GOTO 560
630 CLS:COLOR 15,1,1

```

```

640 PRINT:PRINT"                R A T I O - C
A L C"
650 PRINT" T E R M I N A L  SOFTWARE PUB
LICATIES"
660 PRINT
670 PRINT"copyright 1985 MSX-versie:R-J.
DONKERS"
680 PRINT"^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^
^^^^^^^^^^^^"
690 PRINT:PRINT
700 PRINT" Dit programma verschaft U de
financielekengetallen van Uw bedrijf van
uit de      Balans en Resultatenrekening."
710 PRINT:PRINT" Verstrek de gegevens in
HELE guldens !"
720 LOCATE 0,22:PRINT"          TOETS > RETUR
N < VOOR VERVOLG "
730 Q$=INKEY$:IF Q$="" THEN 730
740 IF ASC(Q$)=13 THEN RETURN
750 GOTO 730
760 N=LEN(X$)
770 I=1
780 IF N>7 THEN PRINT"WAARDE TE GROOT !!
":I=0:GOTO 840
790 FOR Y=1 TO N
800 IF ASC(MID$(X$,Y,1))<48 OR ASC(MID$(
X$,Y,1))>58 THEN I=0
810 NEXT Y
820 IF J=0 THEN 840
830 IF I=0 THEN PRINT"GEEF WAARDE IN HEL
E GULDENS !"
840 RETURN
850 REM *****
      ** VERANDERINGEN AANBRENGEN **
      *****
860 CLS:COLOR 1,3,3
870 J=0
880 PRINT:PRINT"          BEZITTINGEN:"
890 LOCATE 0,4:PRINT" 0 => VERDER GAAN"
900 FOR X=1 TO 5

```

```

910 LOCATE 0,5+X:PRINT X:".>"
920 LOCATE 6,5+X:PRINT A$(X)
930 LOCATE 28,5+X:PRINT USING"f1 #####"
#";A(X)
940 NEXT X
950 LOCATE 0,13:INPUT" WIJZIGEN (0,1,2,3
,4,5)";X$
960 M=VAL(X$)
970 IF M<0 OR M>5 THEN 950
980 IF M=0 THEN 1040
990 LOCATE 0,15:PRINT"GEEF NIEUWE WAARDE
VOOR:";A$(M)
1000 LOCATE 0,17:INPUT X$
1010 GOSUB 760:IF I=0 THEN 1000
1020 A(M)=VAL(X$)
1030 GOTO 850
1040 CLS:COLOR 1,3,3
1050 J=0
1060 PRINT:PRINT"          SCHULDEN:"
1070 LOCATE 0,4:PRINT" 0 => VERDER GAAN"
1080 FOR X=1 TO 3
1090 LOCATE 0,5+X:PRINT X:".>"
1100 LOCATE 6,5+X:PRINT B$(X)
1110 LOCATE 28,5+X:PRINT USING"f1 #####"
##";B(X)
1120 NEXT X
1130 LOCATE 0,13:INPUT" WIJZIGEN (0,1,2,
3)";X$
1140 M=VAL(X$)
1150 IF M<0 OR M>3 THEN 1130
1160 IF M=0 THEN 1220
1170 LOCATE 0,15:PRINT"GEEF NIEUWE WAARD
E VOOR:";B$(M)
1180 LOCATE 0,17:INPUT X$
1190 GOSUB 760:IF I=0 THEN 1170
1200 B(M)=VAL(X$)
1210 GOTO 1040
1220 CLS:COLOR 1,3,3
1230 J=0
1240 PRINT:PRINT"          EIGEN VERMOGEN:"

```

```

1250 LOCATE 0,4:PRINT" 0 => VERDER GAAN"
1260 FOR X=1 TO 4
1270 LOCATE 0,5+X:PRINT X;".">"
1280 LOCATE 6,5+X:PRINT C$(X)
1290 LOCATE 28,5+X:PRINT USING"†1 #####
##";C$(X)
1300 NEXT X
1310 LOCATE 0,13:INPUT" WIJZIGEN (0,1,2,
3,4)";X$
1320 M=VAL(X$)
1330 IF M<0 OR M>4 THEN 1310
1340 IF M=0 THEN 1400
1350 LOCATE 0,15:PRINT"GEEF NIEUWE WAARD
E VOOR:";C$(M)
1360 LOCATE 0,17:INPUT X$
1370 GOSUB 760:IF I=0 THEN 1360
1380 C(M)=VAL(X$)
1390 GOTO 1220
1400 REM *****
** SUBTOTALEN OPNIEUW **
** UITREKENEN **
*****
1410 A=0:B=0:C=0
1420 FOR X=1 TO 5
1430 A=A+A(X)
1440 NEXT X
1450 FOR X=1 TO 3
1460 B=B+B(X)
1470 NEXT X
1480 FOR X=1 TO 4
1490 C=C+C(X)
1500 NEXT X
1510 RETURN
1520 REM *****
**
** H O O F D M E N U **
**
*****
1530 IF I=0 THEN GOSUB 160
1540 CLS:COLOR 15,4,4

```

```

1550 PRINT"                                     *** M E N U ***"
1560 PRINT:PRINT"^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^^"
1570 LOCATE 0,5:PRINT"BALANS opstellen--
-----=> 1"
1580 PRINT:PRINT"VERLIES- en WINSTREKENI
NG=> 2"
1590 PRINT:PRINT"ANALYSEREN KENGETALLEN-
---=> 3"
1600 LOCATE 0,22:PRINT"                                     TOETS U
W KEUZE IN !"
1610 Q$=INKEY$:IF Q$="" THEN 1610
1620 IF Q$<"1" OR Q$>"3" THEN 1610
1630 IF Q$="1" THEN GOSUB 1670:GOTO 1520
1640 IF Q$="2" THEN GOSUB 1910:GOTO 1520
1650 IF Q$="3" THEN GOSUB 2120:GOTO 1520
1660 GOTO 1610
1670 REM *****
      **                                     **
      ** B A L A N S   O P S T E L L E N **
      **                                     **
      *****
1680 CLS:COLOR 12,3,3
1690 PRINT"                                     B A L A N S":PRINT
1700 PRINT"EIGEN VERMOGEN:"
1710 FOR X=1 TO 4
1720 PRINT C$(X);:PRINT TAB(19);USING"f1
#####";C(X)
1730 NEXT X
1740 LOCATE 28,7:PRINT USING"f1 #####
";C
1750 LOCATE 28,8:PRINT"=====
1760 PRINT"BESTAAT UIT      ":"
1770 FOR X=1 TO 5
1780 PRINT A$(X);:PRINT TAB(19);USING"f1
#####";A(X)
1790 NEXT X
1800 LOCATE 28,15:PRINT USING"f1 #####
#":A
1810 LOCATE 28,16:PRINT"=====

```

```

1820 PRINT"VERMINDERD MET:"
1830 FOR X=1 TO 3
1840 PRINT B$(X);:PRINT TAB(19);USING"f1
#####";B(X)
1850 NEXT X
1860 LOCATE 28,21:PRINT USING"f1 #####
#":B
1870 LOCATE 28,22:PRINT"=====
1880 LOCATE 0,22:PRINT">RETURN< VOOR MEN
U"
1890 Q$=INKEY$:IF Q$="" THEN 1890
1900 IF ASC(Q$)=13 THEN RETURN ELSE 1890
1910 REM *****
**
**VERLIES- EN WINSTREKENING**
** OPSTELLEN **
*****
1920 CLS:COLOR 14,1,1
1930 PRINT" VERLIES- & WINSTREKENI
NG"
1940 PRINT" ^^^^^^^^^^^^^^^^^^^^^^^^^^^^^
^^":PRINT
1950 PRINT"ONTVANGSTEN:":PRINT
1960 FOR X=1 TO 3
1970 PRINT" ";D$(X);:PRINT TAB(28) USIN
G"f1 #####";D(X)
1980 NEXT X
1990 PRINT
2000 PRINT"MINUS KOSTEN:":PRINT
2010 FOR X=1 TO 5
2020 PRINT" ";E$(X);:PRINT TAB(28) USIN
G"f1 #####";E(X)
2030 NEXT X
2040 PRINT TAB(28)"-----"
2050 IF D>E THEN PRINT"WINST: ";TAB(28);
2060 IF E>D THEN PRINT"VERLIES: ";TAB(28)
;
2070 PRINT USING"f1 #####";(D-E)
2080 LOCATE 0,22:PRINT" TOETS > RET
URN < VOOR MENU"

```



```

2090 Q$=INKEY$:IF Q$="" THEN 2090
2100 IF ASC(Q$)=13 THEN RETURN
2110 GOTO 2090
2120 REM *****
      **
      **   TABEL VAN KENGETALLEN   **
      **
      *****
2130 CLS:COLOR 11,4,4
2140 PRINT"TABEL FINANCIËLE KENGETALLEN"
2150 PRINT"=====
2160 PRINT
2170 PRINT"CURRENT RATIO           ="
      ;INT(((A-A(5))/(B-B(3))*100+.5)/100
2180 PRINT
2190 PRINT"QUICK RATIO               ="
      ;INT((((A-A(1)-A(5))/(B-B(3))))*100+.5)/
100
2200 PRINT
2210 PRINT"DEKKING SCHULDEN           ="
      ;INT((B/A)*100+.5)/100
2220 PRINT
2230 PRINT"DEKKING RENTELASTEN         ="
      ;INT(((D-E+E(2)+E(4))/E(2))*100+.5)/100
2240 PRINT
2250 PRINT"DEKKING VASTE LASTEN         ="
      ;INT(((D-E(1)-E(5))/(E(2)+E(4)))*100+.5)
/100
2260 PRINT
2270 PRINT"OML.SNELH.VASTG.MIDDELEN    ="
      ;INT(((D(1)+D(2))/A(5))*100+.5)/100
2280 PRINT
2290 PRINT"OML.SNELH.TOTALE ACTIVA      ="
      ;INT(((D(1)+D(2))/A)*100+.5)/100
2300 PRINT
2310 PRINT"WINSTMARGE                     ="
      ;INT(((D-E)/(D(1)+D(2)))*100+.5)/100
2320 PRINT
2330 PRINT"RENTABILITEIT V/H BEDRIJF    ="
      ;INT(((D-E)/A)*100+.5)/100

```

```

2340 LOCATE 0,22:PRINT" TOETS RETURN VO
OR MENU "
2350 Q$=INKEY$:IF Q$="" THEN 2350
2360 IF ASC(Q$)=13 THEN RETURN
2370 GOTO 2350
2380 DATA "Voorraad"."Liq.middelen","Vor
deringen","Over.Vlot.Middelen","Vastgel.
Middelen"
2390 DATA "Bankleningen","Schulden Kort"
,"Schulden Lang","Kapitaal","Reserves","
Onverd.Winst","Overige"
2400 DATA "Verkoop Contant","Verkoop op
Rek.","Overige Inkomsten","Inkopen/Afsch
r.","Rente","Huur"
2410 DATA "Belasting","Overige Bedrijfsk
ost."

```

AANTEKENINGEN: